

Learn Real World Haskell

Jacob Bishop

2023-08-20

1 Getting started

略

2 Types and functions

略

3 Defining types, streamlining functions

略

4 Functional programming

略

5 Writing a library: working with JSON data

在 Haskell 中表示 JSON

首先是在 Haskell 中定义 JSON 这个数据，这里使用代数数据类型来表达 JSON 类型的范围。

```
1 data JValue
2   = JString String
3   | JNumber Double
4   | JBool Bool
5   | JNull
6   | JObject [(String, JValue)]
7   | JArray [JValue]
8   deriving (Eq, Ord, Show)
```

对于每种 JSON 类型，我们都提供了独立的值构造函数。测试：

```
1 ghci> :l SimpleJSON
2 [1 of 1] Compiling Main             ( SimpleJSON.hs, interpreted )
3 Ok, one module loaded.
4 ghci> JString "foo"
5 JString "foo"
6 ghci> JNumber 2.7
7 JNumber 2.7
8 ghci> :type JBool True
9 JBool True :: JValue
```

构造一个从 `JValue` 获取字符串的函数：

```
1 getString :: JValue -> Maybe String
2 getString (JString s) = Just s
3 getString _ = Nothing
```

测试：

```
1 ghci> :reload
2 [1 of 1] Compiling Main             ( SimpleJSON.hs, interpreted )
3 Ok, one module loaded.
4 ghci> getString (JString "hello")
5 Just "hello"
6 ghci> getString (JNumber 3)
7 Nothing
```

接下来是其它类型的访问函数：

```
1 getInt (JNumber n) = Just n
2 getInt _ = Nothing
3
4 getDouble (JNumber n) = Just n
5 getDouble _ = Nothing
```

```

6
7  getBool (JBool n) = Just n
8  getBool _ = Nothing
9
10 getObject (JObject o) = Just o
11 getObject _ = Nothing
12
13 getArray (JArray a) = Just a
14 getArray _ = Nothing
15
16 isNull v = v == JNull

```

`truncate` 函数可以让浮点类型或者有理数去掉小数点后变为整数：

```

1 ghci> truncate 5.8
2 5
3 ghci> :module +Data.Ratio
4 ghci> truncate (22 % 7)
5 3

```

Haskell 模块详解

一个 Haskell 源文件包含了单个模块的定义。模块允许我们在它其内部进行定义，并由其它模块访问：

```

1 module SimpleJSON
2   ( JValue (..),
3     getString,
4     getInt,
5     getDouble,
6     getBool,
7     getObject,
8     getArray,
9     isNull,
10  )
11 where

```

如果省略了导出（即圆括号以及其所包含的名称），那么该模块中的所有名称都会被导出。

编译 Haskell 源

编译一个源文件：

```

1 ghc -c SimpleJSON.hs

```

`-c` 选项告诉 `ghc` 仅生成对象代码。如果省略了该选项，那么编译器则会尝试生成一整个可执行文件。这会导致失败，因为我们并没有一个 `main` 函数，即 GHC 所认为的一个独立程序的执行入口。

编译后会得到两个新文件：`SimpleJSON.hi` 与 `SimpleJSON.o`。前者是一个接口 *interface* 文件，即 `ghc` 以机器码的形式存储模块导出的名称信息；后者是一个对象 *object* 文件，其包含了生产的机器码。

生成一个 Haskell 程序，导入模块

添加一个 `Main.hs` 文件，其内容如下：

```
1 module Main where
2
3 import SimpleJSON
4
5 main = print (JObject [("foo", JNumber 1), ("bar", JBool False)])
```

与原文 `module Main () where` 的不同之处在于，现在的 `Main` 后不再需要一个 `()`。接下来是编译 `main` 函数：

```
1 ghc -o simple Main.hs
```

与原文 `ghc -o simple Main.hs SimpleJSON.o` 不同，现在没了 `SimpleJSON.o` 这个文件，加上后会报错重复 symbol 的编译错误（因为在 `Main.hs` 中已经做了 `import SimpleJSON` 导入了）。

这次省略掉了 `-c` 选项，因此编译器尝试生成一个可执行。生成可执行的过程被称为链接 *linking*（与 C++ 一样），即在一次编译中链接源文件与可执行文件。

这里给了 `ghc` 一个新选项 `-o`，其接受一个参数：可执行文件的名称，这里是 `simple`，执行它：

```
1 ./simple
2 JObject [("foo",JNumber 1.0),("bar",JBool False)]
```

打印 JSON 数据

现在我们将 Haskell 的值渲染成 JSON 数据，创建一个 `PutJSON.hs` 文件：

```
1 module PutJSON where
2
3 import Data.List (intercalate)
4 import SimpleJSON
5
6 renderJValue :: JValue -> String
7 renderJValue (JString s) = show s
8 renderJValue (JNumber n) = show n
9 renderJValue (JBool True) = "true"
10 renderJValue (JBool False) = "false"
11 renderJValue JNull = "null"
12 renderJValue (JObject o) = "{" ++ pairs o ++ "}"
```

```

13     where
14         pairs [] = ""
15         pairs ps = intercalate ", " (map renderPair ps)
16         renderPair (k, v) = show k ++ ": " ++ renderJValue v
17     renderJValue (JArray a) = "[" ++ values a ++ "]"
18     where
19         values [] = ""
20         values vs = intercalate ", " (map renderJValue vs)

```

好的 Haskell 风格需要分隔纯代码与 I/O 代码。我们的 `renderJValue` 函数不会与外界交互，但是仍然需要一个打印的函数：

```

1 putJValue :: JValue -> IO ()
2 putJValue = putStrLn . renderJValue

```

类型推导是把双刃剑

假设我们编写了一个自认为返回 `String` 的函数，但是并不为其写类型签名：

```

1 upcaseFirst (c:cs) = toUpper c -- forgot ":cs" here

```

这里希望单词首字母大写，但是忘记了将剩余的字符放进结果中。我们认为函数的类型是 `String -> String`，但是编译器则会将其视为 `String -> Char`。假设我们尝试在其他地方调用该函数：

```

1 camelCase :: String -> String
2 camelCase xs = concat (map upcaseFirst (words xs))

```

那么当我们尝试编译该代码或者是加载进 `ghci`，我们并不会得到明显的错误信息：

```

1 ghci> :load Trouble
2 [1 of 1] Compiling Main                ( Trouble.hs, interpreted )
3
4 Trouble.hs:9:27:
5   Couldn't match expected type `[Char]' against inferred type `Char'
6     Expected type: [Char] -> [Char]
7     Inferred type: [Char] -> Char
8     In the first argument of `map', namely `upcaseFirst'
9     In the first argument of `concat', namely
10        `(map upcaseFirst (words xs))'
11 Failed, modules loaded: none.

```

注意这里的报错是在 `upcaseFirst` 函数处，那么假设我们认为 `upcaseFirst` 的定义与类型是正确的，那么查找到真正的错误可能会花掉一些时间。

更加泛用的渲染

我们将更为泛用的打印模块称为 `Prettify`，那么其源文件即 `Prettify.hs`。

为了使 `Prettify` 满足实际需求，我们还要一个新的 JSON 渲染器来使用 `Prettify` 的 API。在 `Prettify` 模块中将使用一个抽象类型 `Doc`。基于建立在抽象类型的泛用渲染库，我们可以选择灵活高效的实现。

`PrettyJSON.hs` 示例：

```
1 renderJValue :: JValue -> Doc
2 renderJValue (JBool True) = text "true"
3 renderJValue (JBool False) = text "false"
4 renderJValue JNull = text "null"
5 renderJValue (JNumber num) = double num
6 renderJValue (JString str) = string str
```

这里的 `text`，`double` 以及 `string` 都会在 `Prettify` 模块中提供。

开发 Haskell 代码而不发疯

一个用于快速开发程序框架的技巧就是编写占位符，或者类型与函数的根 *stub* 版本。例如上述 `string`，`text` 以及 `double` 函数将被 `Prettify` 模块提供。如果我们没有提供这些函数或者 `Doc` 类型，那么“早点编译，经常编译”这个尝试就会失败。为了避免这个问题现在让我们编写一个不做任何事情的程序。

```
1 import SimpleJSON
2
3 data Doc = ToBeDefined deriving (Show)
4
5 string :: String -> Doc
6 string str = undefined
7
8 text :: String -> Doc
9 text str = undefined
10
11 double :: Double -> Doc
12 double num = undefined
```

特殊值 `undefined` 有一个 `a` 类型，无论在哪里使用它，总是会有类型检查。如果尝试计算，则会使程序崩溃：

```
1 ghci> :type undefined
2 undefined :: a
3 ghci> undefined
4 *** Exception: Prelude.undefined
5 ghci> :type double
6 double :: Double -> Doc
7 ghci> double 3.14
8 *** Exception: Prelude.undefined
```

尽管还不能运行根代码，但是编译器的类型检查器可以确保我们的程序类型正确。

漂亮的打印字符串

当需要打印一个漂亮的字符串时，我们必须遵循 JSON 的转义规则。字符串就是一系列被包裹在引号中的字符们。 `PrettyJSON.hs`：

```
1 string :: String -> Doc
2 string = enclose '"' ' ' . hcat . map oneChar
```

以及 `enclose` 函数将一个 `Doc` 值简单的包裹在一个开始与结束字符之间：

```
1 enclose :: Char -> Char -> Doc -> Doc
2 enclose left right x = char left <> x <> char right
```

这里提供的 `<>` 函数定义在 `Prettify` 库中，它需要两个 `Doc` 值，即 `Doc` 版本的 `(++)`。还是先在根文件中定义：

```
1 (<>) :: Doc -> Doc -> Doc
2 a <> b = undefined
3
4 char :: Char -> Doc
5 char c = undefined
```

我们的库还需要提供 `hcat`，将若干 `Doc` 值合成成为一个（类似于列表的 `concat`）：

```
1 hcat :: [Doc] -> Doc
2 hcat xs = undefined
```

我们的 `string` 函数应用 `oneChar` 函数到字符串中的每个字符，将它们连接，然后用引号包装。而 `oneChar` 函数则是用来转义或包装一个独立的字符。在 `PrettyJSON.hs` 中：

```
1 oneChar :: Char -> Doc
2 oneChar c = case lookup c simpleEscapes of
3   Just r -> text r
4   Nothing
5     | mustEscape c -> hexEscape c
6     | otherwise -> char c
7 where
8   mustEscape c = c < ' ' || c == '\x7f' || c > '\xff'
9
10 simpleEscapes :: [(Char, String)]
11 simpleEscapes = zipWith ch "\b\n\f\r\t\\\"/" "bnfrt\\\"/"
12 where
13   ch a b = (a, ['\\', b])
```

这里的 `simpleEscapes` 是一个列表的二元组，我们称其为关联 *association* 列表，或者简称 `alist`。每个 `alist` 的元素都将字符关联了它的转义表达，测试：

```
1 ghci> take 4 simpleEscapes
2 [('b', "\\b"), ('n', "\\n"), ('f', "\\f"), ('r', "\\r")]
```


我们的 `case` 表达式尝试查看字符是否匹配关联列表。如果匹配则返回匹配值，如若不然则需要以更复杂的方式来转义该字符。只有当两种转义都不需要时，才会返回普通字符。保守起见，我们输出的唯一未转义字符是可打印的 ASCII 字符。

更复杂的转义包含了将一个字符转为字符串 `"\u"` 并跟随四个十六进制的字符用于表达 Unicode 字符的数值。仍然是 `PrettyJSON.hs`：

```
1 smallHex :: Int -> Doc
2 smallHex x =
3   text "\\u"
4   <> text (replicate (4 - length h) '0')
5   <> text h
6   where
7     h = showHex x ""
```

这里的 `showHex` 函数需要从 `Numeric` 库中加载，其用于返回一个值的十六进制：

```
1 ghci> showHex 114111 ""
2 "1bdbf"
```

`replicate` 函数则是由 `Prelude` 提供：

```
1 ghci> replicate 5 "foo"
2 ["foo","foo","foo","foo","foo"]
```

`smallHex` 提供的四位编码只能表示最大 `0xffff` 的 Unicode 字符，而有效的 Unicode 字符的范围可以达到 `0x10ffff`。为了正确的表达一个超出了 `0xffff` 的 JSON 字符串，我们遵循一些复杂的规则将其分为两部分。这使我们有机会对 Haskell 的数执行一些位级操作。还是 `PrettyJSON.hs`：

```
1 astral :: Int -> Doc
2 astral n = smallHex (a + 0xd800) <> smallHex (b + 0xdc00)
3   where
4     a = (n `shiftR` 10) .&. 0x3ff
5     b = n .&. 0x3ff
```

这里 `shiftR` 函数和 `(.&.)` 函数都源自 `Data.Bits` 模块，前者将一个数移动右一位，后者则是执行一个字节层面的两值 *and* 操作。

```
1 ghci> 0x10000 `shiftR` 4    :: Int
2 4096
3 ghci> 7 .&. 2              :: Int
4 2
```

现在有了 `smallHex` 与 `astral`，我们可以提供 `hexEscape` 的定义了：

```
1 hexEscape :: Char -> Doc
2 hexEscape c
3   | d < 0x10000 = smallHex d
4   | otherwise = astral (d - 0x10000)
```

```

5  where
6    d = ord c

```

其中 `ord` 由 `Data.Char` 模块提供。

数组与对象，以及模块头

相比于字符串的漂亮打印，数组和对象就是小菜一碟了。我们已经知道了它们两者其实很相似：都是由起始字符开始，接着是一系列由逗号分隔的值，最后接上结束字符。让我们编写一个函数捕获数组与对象的共同结构：

```

1  series :: Char -> Char -> (a -> Doc) -> [a] -> Doc
2  series open close item = enclose open close . fsep . punctuate (char ',') . map item

```

让我们首先从函数类型开始。它接受起始与结束字符，一个打印某些未知类型 `a` 值的函数，以及一个类型为 `a` 的列表，返回一个类型为 `Doc` 的值。

注意尽管我们的类型签名提及了四个参数，在函数定义中仅列出了三个。这遵循了简化定义的规则，如 `myLength xs = length xs` 等同于 `myLength = length`。

我们已经有了之前编写过的 `enclose`，即包装一个 `Doc` 值进起始与结束字符之间。那么 `fsep` 函数则位于 `Prettify` 模块中，其结合一个 `Doc` 值列表成为一个 `Doc`，在输出不适合单行的情况下还需要换行。

```

1  fsep :: [Doc] -> Doc
2  fsep xs = undefined

```

那么现在，遵循上述提供的例子，你应该能够定义你自己的 `Prettify.hs` 的根文件了。这里不再显式的定义更多的根了。

`punctuate` 函数同样位于 `Prettify` 模块中：

```

1  punctuate :: Doc -> [Doc] -> [Doc]
2  punctuate p [] = []
3  punctuate p [d] = [d]
4  punctuate p (d : ds) = (d <> p) : punctuate p ds

```

通过 `series` 的定义，漂亮打印一个数组就非常直接了：

```

1  renderJValue (JArray ary) = series '[' ']' renderJValue ary

```

对于对象而言，还需要额外的一些工作：每个元素同时需要处理名称与值：

```

1  renderJValue (JObject obj) = series '{' '}' field obj
2  where
3    field (name, val) =
4      string name
5        PrettyStub.<> text ":"
6        PrettyStub.<> renderJValue val

```

编写一个模块头

现在已经有了 `PrettyJSON.hs` 文件，我们需要回到其顶部添加模块声明：

```
1 module PrettyJSON (renderJValue) where
```

这里导出了一个名称：`renderJValue`，也就是我们的 JSON 渲染函数。模块中其它的定义都是用于支持 `renderJValue` 的，因此没有必要对其它模块可见。

充实我们的漂亮打印库

在 `Prettify` 模块中，提供了 `Doc` 类型作为一个代数数据类型：

```
1 data Doc
2   = Empty
3   | Char Char
4   | Text String
5   | Line
6   | Concat Doc Doc
7   | Union Doc Doc
8   deriving (Show)
```

观察可知 `Doc` 类型实际上是一颗树。`Concat` 与 `Union` 构造函数根据其它两个 `Doc` 值创建一个内部节点，而 `Empty` 以及其它简单的构造函数用于构建叶子。

在模块的头部，我们导出该类型的名称，而不是它们的构造函数：这样可以防止使用 `Doc` 的构造函数被用于创建与模式匹配 `Doc` 值。

相反的，要创建一个 `Doc`，用户需要调用我们在 `Prettify` 模块中所提供的函数：

```
1 empty :: Doc
2 empty = Empty
3
4 char :: Char -> Doc
5 char = Char
6
7 text :: String -> Doc
8 text "" = Empty
9 text s = Text s
10
11 double :: Double -> Doc
12 double = text . show
```

`Line` 构造函数代表一个换行，其创建的是一个 *hard* 换行，即总是会在漂亮的打印中出现。有时我们想要一个 *soft* 换行，即只会在窗口或者页面上过长显示时才会换行。稍后将会介绍 `softline` 函数。

```
1 line :: Doc
2 line = Line
```

另外就是用于连接两个 `Doc` 值的 `(<>)` 函数（这里使用 `(<+>)`，因为 `(<>)` 在 Prelude 中已经有定义了）：

```
1 (<+>) :: Doc -> Doc -> Doc
2 Empty <+> y = y
3 x <+> Empty = x
4 x <+> y = x `Concat` y
```

测试：

```
1 ghci> text "foo" <> text "bar"
2 Concat (Text "foo") (Text "bar")
3 ghci> text "foo" <> empty
4 Text "foo"
5 ghci> empty <> text "bar"
6 Text "bar"
```

接下来是用于连接 `Doc` 列表的 `hcat` 函数：

```
1 hcat :: [Doc] -> Doc
2 hcat = fold (<+>)
3
4 fold :: (Doc -> Doc -> Doc) -> [Doc] -> Doc
5 fold f = foldr f empty
```

以及 `fsep` 函数，它还依赖若干其它函数：

```
1 fsep :: [Doc] -> Doc
2 fsep = fold (</>)
3
4 (</>) :: Doc -> Doc -> Doc
5 x </> y = x <+> softline <+> y
6
7 softline :: Doc
8 softline = group line
```

这里需要解释一下，`softline` 函数应该在当前行特别宽的时候进行换行，否则插入空格。如果我们的 `Doc` 类型不包含任何关于渲染的信息，那么该如何呢？答案就是每次遇到一个软换行，通过 `Union` 构造函数来维护两个可选项：

```
1 group :: Doc -> Doc
2 group x = flatten x `Union` x
```

`flatten` 函数将一个 `Line` 替换为空格，将两行转换为一个更长的行。

```
1 flatten :: Doc -> Doc
2 flatten (x `Concat` y) = flatten x `Concat` flatten y
3 flatten Line = Char ' '
4 flatten (x `Union` _) = flatten x
5 flatten other = other
```

注意总是调用 `flatten` 在 `Union` 的左元素上：每个 `Union` 的左侧总是大于等于右侧宽度（字符距离）。

紧密渲染

我们需要频繁的使用包含尽可能少字符的数据。例如通过网络连接发送 JSON 数据就没有必要美观：软件并不在乎数据的美观与否，添加很多空格只会带来性能下降。

因此我们提供了一个紧密渲染的函数：

```
1 compact :: Doc -> String
2 compact x = transform [x]
3 where
4     transform [] = ""
5     transform (d : ds) = case d of
6         Empty -> transform ds
7         Char c -> c : transform ds
8         Text s -> s ++ transform ds
9         Line -> '\n' : transform ds
10        a `Concat` b -> transform (a : b : ds)
11        _ `Union` b -> transform (b : ds)
```

`compact` 函数将其参数包裹成一个列表，然后将帮助函数 `transform` 应用至该列表。`transform` 函数视其参数为堆叠的项用于处理，列表的第一个元素即堆的顶部。

`transform` 函数的 `(d:ds)` 模式将堆顶部的元素取出 `d` 并留下 `ds`。在 `case` 表达式中，前面几个分支都在 `ds` 上递归，每次递归消费堆顶部的元素；最后两个分支则是在 `ds` 之前添加项：`Concat` 添加两个元素至堆，而 `Union` 分支则忽略它左侧元素，调用的是 `flatten`，再将其右侧元素至堆。

测试 `compact`：

```
1 ghci> let value = renderJValue (JObject [("f", JNumber 1), ("q", JBool True)])
2 ghci> :type value
3 value :: Doc
4 ghci> putStrLn (compact value)
5 {"f": 1.0,
6  "q": true
7 }
```

为了更好的理解代码是如何运作的，让我们用一个更简单的例子来展示细节：

```
1 ghci> char 'f' <> text "oo"
2 Concat (Char 'f') (Text "oo")
3 ghci> compact (char 'f' <> text "oo")
4 "foo"
```

当我们应用 `compact` 时，它会将它的参数转为一个列表后应用函数 `transform`。

- 接下来 `transform` 函数接受了一个单例列表，然后进行模式匹配 `(d:ds)`，这里的 `d` 是 `Concat (Char 'f') (Text "oo")`，而 `ds` 则是一个空列表。

由于 `d` 的构造函数是 `Concat`，其模式匹配就在 `case` 表达式中。那么在右侧，添加 `Char 'f'` 与 `Text "oo"` 值堆，然后递归的应用 `transform`。

- `transform` 函数接受了一个包含两项的列表，继续匹配 `d:ds` 模式。此时变量 `d` 绑定到了 `Char 'f'`，而 `ds` 则是 `[Text "oo"]`。
`case` 表达式匹配到了 `Char` 分支。那么在右侧，使用 `(:)` 来构建一个列表，其头部为 `'f'`，其余部分则是递归应用 `transform` 后的结果。
- * 递归的调用接受到一个单例列表，其变量 `d` 绑定至 `Text "oo"`，`ds` 绑定至 `[]`。
`case` 表达式匹配 `Text` 分支。那么在右侧，使用 `(++)` 来连接 `"oo"` 与递归调用 `transform` 后的结果。
 * · 最后的调用，`transform` 得到一个空列表，返回一个空字符串。
 * 结果是 `"oo" ++ ""`
- 结果是 `'f' : "oo" ++ ""`

真实的漂亮打印

我们的 `compact` 函数对于机器之间的交流是有帮助的，但是它的结果对人类而言并不易读：每行的信息很少。为了生成一个更可读的输出，我们将要编写另一个函数 `pretty`。相比于 `compact`，`pretty` 接受一个额外的参数：一行的最大宽度。

```
1 pretty :: Int -> Doc -> String
```

确切来说，`Int` 参数控制了 `pretty` 在遇到一个 `softline` 时的行为。在一个 `softline` 时，`pretty` 会选择继续留在当前行还是另起一行。其余情况下，必须严格遵守漂亮打印函数所设定的指令。

以下是实现的代码：

```
1 pretty width x = best 0 [x]
2 where
3   best col (d : ds) = case d of
4     Empty -> best col ds
5     Char c -> c : best (col + 1) ds
6     Text s -> s ++ best (col + length s) ds
7     Line -> '\n' : best 0 ds
8     a `Concat` b -> best col (a : b : ds)
9     a `Union` b -> nicest col (best col (a : ds)) (best col (b : ds))
10  best _ _ = ""
11  nicest col a b
12    | (width - least) `fits` a = a
13    | otherwise = b
14  where
15    least = min width col
```

`best` 帮助函数接受两个参数：当前行使用了的列数，以及剩余的仍需处理的 `Doc` 列表。

在简单的情况下，随着消费输入 `best` 直接更新了 `col` 变量。其中 `Concat` 情况也很明显：将两个连接过的部分推至堆叠，且不触碰 `col`。

有趣的情况在 `Union` 构造函数。回想一下之前将 `flatten` 应用至左侧元素，且不对右侧做任何操作。还有就是 `flatten` 将新行替换成空格。因此我们则需要检查这两种布局，`flatten` 后的还是原始的，更适合我们的宽度限制。

为此需要编写一个小的帮助函数来决定 `Doc` 值的一行是否适合给定的长度：

```
1 fits :: Int -> String -> Bool
2 w `fits` _ | w < 0 = False
3 w `fits` "" = True
4 w `fits` ('\n' : _) = True
5 w `fits` (c : cs) = (w - 1) `fits` cs
```

遵循漂亮打印

为了理解代码如何工作的，首先考虑一个简单的 `Doc` 值：

```
1 ghci> empty </> char 'a'
2 Concat (Union (Char ' ') Line) (Char 'a')
```

我们将应用 `pretty 2` 在该值。当我们首先应用 `best`，`col` 值为零。它匹配 `Concat` 模式，接着将 `Union (Char ' ') Line` 与 `Char 'a'` 推至堆，接着是递归应用自身，它匹配了 `Union (Char ' ') Line`。

现在忽略 Haskell 通常的计算顺序，两个子表达式，`best 0 [Char ' ', char 'a']` 与 `best 0 [Line, Char 'a']`，前者计算得到 `" a"`，而后者得到 `"na "`。接着将它们替换到外层的表达式中，得到 `nicest 0 " a" "\na"`。

为了明白 `nicest` 的结果，我们做一个小小的替换。`width` 以及 `col` 分别是 0 与 2，那么 `least` 就是 0，`width - least` 就是 2。这里用 `ghci` 来计算一下 `2 `fits` " a"`：

```
1 ghci> 2 `fits` " a"
2 True
```

计算得到 `True`，那么这里的 `nicest` 结果就是 `" a"`。

如果将 `pretty` 函数应用到之前同样的 JSON 数据上，可以看到根据提供的最大长度，它会得到不同的结果：

```
1 ghci> putStrLn (pretty 10 value)
2 {"f": 1.0,
3  "q": true
4  }
5 ghci> putStrLn (pretty 20 value)
6 {"f": 1.0, "q": true
7  }
8 ghci> putStrLn (pretty 30 value)
9 {"f": 1.0, "q": true }
```

创建一个库

Haskell 社区构建了一个标准工具库，名为 Cabal，其用于构建，安装，以及分发软件。Cabal 以 `包package` 的方式管理软件，一个包包含了一个库，以及若干可执行程序。

编写一个包的描述

要使用包，Cabal 需要一些描述。这些描述保存在一个文本文件中，以 `.cabal` 后缀命名。该文件应位于项目的根目录。

包描述由一系列的全局属性开始，其应用于包中所有的库以及可执行。

```
1 name:           pretty-json
2 version:        0.1.0.0
```

包名称必须是唯一的。如果你创建并安装了一个同名包在你的系统上，GHC 会感到迷惑。

```
1 synopsis:       My pretty printing library, with JSON support
2 description:
3   A simple pretty printing library that illustrates how to
4   develop a Haskell library.
5 author:         jacob xie
6 maintainer:     jacobbishopxy@gmail.com
```

这里还要有 license 信息，大多数 Haskell 包都使用 BSD license，Cabal 称为 BSD3。

另外就是 Cabal 的版本：

```
1 cabal-version:  2.4
```

在一个包中要描述一个独立的库，需要 *library* 这个部分。注意缩进在这里很重要。

```
1 library
2   default-language: Haskell2010
3   build-depends:    base
4   exposed-modules:
5     Prettify
6     PrettyJSON
7     SimpleJSON
```

`exposed-modules` 字段包含了一个模块列表，用于暴露给使用该包的用户导入。另一个可选字段 `other-modules` 包含了一个内部 *internal* 模块的列表，它们提供给 `exposed-modules` 内的模块使用，而对用户不可见。

`build-depends` 字段包含了一个逗号分隔的包列表，它们是我们库所需的依赖。`base` 包中包含了 Haskell 很多核心模块，例如 Prelude，所以它总是必须的。

GHC 的包管理器

GHC 包含了一个简单的包管理器用于追踪安装了哪些包，以及这些包的版本号。一个名为 `ghc-pkg` 的命令行工具提供了包数据库的管理。

这里说数据库 *database* 是因为 GHC 区分了系统级别的包，即对所有用户可用；以及用户可见的包，即仅对当前用户可用。后者可以避免管理员权限来安装包。

ghc-pkg 提供了不同的子命令，多数时候我们仅需两个命令：**ghc-pkg list** 用于查看已安装的包；当需要删除包时则使用 **ghc-pkg unregister**。

设置，构建与安装

除了一个 **.cabal** 文件，一个包还必须包含一个 *setup* 文件。在包需要的情况下，它允许 Cabal 的构建过程中包含大量的定制。

Setup.hs 示例：

```
1  #!/usr/bin/env runhaskell
2
3  import Distribution.Simple
4
5  main = defaultMain
```

一旦有了 **.cabal** 与 **Setup.hs** 文件，那么就剩下三步了。

指导 Cabal 如何构建以及在哪里安装包，只需一个简单命令：

```
1  runghc Setup configure
```

这可以确保我们所需要的包都是可用的，并且保存设定为了之后的 Cabal 命令。

如果没有为 **configure** 提供任何参数，Cabal 则会将包安装在系统层的包数据库。要在 home 路径安装则需要提供一些额外的信息：

```
1  runghc Setup configure --prefix=$HOME --user
```

接下来就是包的构建：

```
1  runghc Setup build
```

如果成功了，我们就可以安装这个包了。我们无需指定其安装的位置：Cabal 会使用我们在 **configure** 步骤中所提供的配置。即安装在我们自己的路径，并更新 GHC 的用于层包数据库。

```
1  runghc Setup install
```

6 Using typeclasses

Typeclasses 是 Haskell 中最强大的特性。它们允许我们定义通用性的接口，为各种类型提供公共特性集。Typeclasses 是一些语言特性的核心，例如相等性测试和数字运算符。

对 Typeclasses 的需求

假设在没有相等性测试 `==` 的情况下需要构建一个简单的 `color` 类型，那么相等测试就应该如下：

```
1 data Color = Red | Green | Blue
2
3 colorEq :: Color -> Color -> Bool
4 colorEq Red Red = True
5 colorEq Green Green = True
6 colorEq Blue Blue = True
7 colorEq _ _ = False
```

现在假设我们想为 `StringS` 增加一个相等性测试。由于 Haskell 的 `String` 是字符列表，我们可以编写一个简单的函数用于测试。这里为了简化我们使用 `==` 操作符用于说明。

```
1 stringEq :: [Char] -> [Char] -> Bool
2 stringEq [] [] = True
3 stringEq (x:xs) (y:ys) = x == y && stringEq xs ys
4 stringEq _ _ = False
```

现在已经发现问题了：我们必须为每个不同的类型使用不同名称的比较函数，这是很低效且令人讨厌的。因此需要一个通用的函数可用于比较任何东西。此外，当新的数据类型之后被创建时，已经存在的代码不能被改变。

Haskell 的 typeclasses 就是设计用来解决上述问题的。

什么是 typeclasses

Typeclasses 定义了一系列的函数，它们可以根据给定的数据类型有不同的实现。

首先我们必须定义 typeclass 本身。我们希望一个函数接受同样类型的两个参数，返回一个 `Bool` 来表示它们是否相等。我们无需在意类型是什么，但需要两个参数的类型相同。下面是 typeclass 的第一个定义：

```
1 class BasicEq a where
2   isEqual :: a -> a -> Bool
```

通过 `ghci` 的类型检查 `:type` 可以得知 `isEqual` 的类型：

```
1 ghci> :type isEqual
2 isEqual :: (BasicEq a) => a -> a -> Bool
```

现在可以为特定类型定义 `isEqual` :

```
1 instance BasicEq Bool where
2   isEqual True True = True
3   isEqual False False = True
4   isEqual _ _ = False
```

测试:

```
1 ghci> isEqual False False
2 True
3 ghci> isEqual False True
4 False
5 ghci> isEqual "Hi" "Hi"
6
7 <interactive>:1:0:
8   No instance for (BasicEq [Char])
9     arising from a use of `isEqual' at <interactive>:1:0-16
10  Possible fix: add an instance declaration for (BasicEq [Char])
11  In the expression: isEqual "Hi" "Hi"
12  In the definition of `it': it = isEqual "Hi" "Hi"
```

注意在尝试比较两个字符串时, `ghci` 发现我们并未给 `String` 提供 `BasicEq` 的实例。因此 `ghci` 并不知道该如何对 `String` 进行比较, 同时提议我们可以通过为 `[Char]` 定义 `BasicEq` 实例来解决这个问题。

下面是定义一个包含了两个函数的 `typeclass`:

```
1 class BasicEq2 a where
2   isEqual2 :: a -> a -> Bool
3   isNotEqual2 :: a -> a -> Bool
```

虽然 `BasicEq2` 的定义没有问题, 但是它让我们做了额外的事情。就逻辑而言, 如果我们知道了 `isEqual` 或 `isNotEqual` 中的一个, 我们便知道了另一个。那么与其让用户定义 `typeclass` 中两个函数, 我们可以提供一个默认的实践。

```
1 class BasicEq3 a where
2   isEqual3 :: a -> a -> Bool
3   isEqual3 x y = not (isNotEqual3 x y)
4
5   isNotEqual3 :: a -> a -> Bool
6   isNotEqual3 x y = not (isEqual3 x y)
```

声明 `typeclass` 实例

现在知道了如何定义 `typeclasses`, 接下来就是直到如何定义 `typeclasses` 的实例。回忆一下, 类型是由一个个特定 `typeclass` 所构成的实例所赋予了意义, 这些 `typeclasses` 又实现了必要的函数。

之前为我们的 `Color` 类型创建了相等性测试，现在让我们试试让其成为 `BasicEq3` 的实例：

```
1 instance BasicEq3 Color where
2   isEqual3 Red Red = True
3   isEqual3 Green Green = True
4   isEqual3 Blue Blue = True
5   isEqual3 _ _ = False
```

注意这里提供了与之前定义的一样的函数，实际上实现是完全相同的。不过在这种情况下，我们可以使用 `isEqual3` 在任何定义了 `BasicEq3` 实例的类型上。

另外注意 `BasicEq3` 定义了 `isEqual3` 与 `isNotEqual3`，而我们仅实现了它们其中一个。这是因为 `BasicEq3` 包含了默认实现，因此没有显式定义 `isNotEqual3` 时，编译器自动的使用了 `BasicEq3` 中的默认实现。

重要的内建 Typeclasses

Show

`Show` typeclass 用于将值转换为 `String`。可能最常用于将数字转换为字符串，很多类型都有它的实例，因此还可以用于转换更多的类型。如果自定义类型实现了 `Show` 的实例，那么就可以在 `ghci` 上展示或者在程序中打印出来。

`Show` 中最重要的函数就是 `show`，它接受一个参数：用于转换的数据，返回一个 `String` 来表示该数据。

```
1 ghci> :type show
2 show :: (Show a) => a -> String
```

一些其它的例子：

```
1 ghci> show 1
2 "1"
3 ghci> show [1, 2, 3]
4 "[1,2,3]"
5 ghci> show (1, 2)
6 "(1,2)"
```

`ghci` 展示的结果与输入到 Haskell 程序中的结果相同。表达式 `show 1` 返回的是单个字符的字符串，其包含数字 `1`，也就是说引号并不是字符串本身。通过 `putStrLn` 可以更清楚的看到：

```
1 ghci> putStrLn (show 1)
2 1
3 ghci> putStrLn (show [1,2,3])
4 [1,2,3]
```

我们也可以使用 `show` 在字符串上：

```

1 ghci> show "Hello!"
2 "\"Hello!\""
3 ghci> putStrLn (show "Hello!")
4 "Hello!"
5 ghci> show ['H', 'i']
6 "\"Hi\""
7 ghci> putStrLn (show "Hi")
8 "Hi"
9 ghci> show "Hi, \"Jane\""
10 "\"Hi, \\\"Jane\\\"\""
11 ghci> putStrLn (show "Hi, \"Jane\"")
12 "Hi, \"Jane\""

```

现在为我们自己的类型实现 `Show` 的实例：

```

1 instance Show Color where
2   show Red = "Red"
3   show Green = "Green"
4   show Blue = "Blue"

```

Read

`Read` typeclass 基本就是相反的 `Show`：它定义了函数接受一个 `String`，分析它，并返回属于 `Read` 成员的任何类型的数据。

```

1 ghci> :type read
2 read :: (Read a) => String -> a

```

以下是一个 `read` 与 `show` 的例子：

```

1 main = do
2   putStrLn "Please enter a Double:"
3   inpStr <- getLine
4   let inpDouble = read inpStr :: Double
5   putStrLn $ "Twice" ++ show inpDouble ++ " is " ++ show (inpDouble * 2)

```

`read` 的类型：`(Read a) => String -> a`。这里的 `a` 是每个 `Read` 实例的类型。也就是说特定的解析函数是根据预期的 `read` 返回类型所决定的。

```

1 ghci> (read "5.0")::Double
2 5.0
3 ghci> (read "5.0")::Integer
4 *** Exception: Prelude.read: no parse

```

在尝试解析 `5.0` 为 `Integer` 时异常。当预期返回值的类型是 `Integer` 时，`Integer` 的解析函数并不接受小数，因此异常被抛出。

`Read` 提供了一些相当复杂的解析器。你可以通过实现 `readsPrec` 函数定义一个简单的解析。该实现在解析成功时，返回只包含一个元组的列表，如果解析失败则返回空列表。下面是一个实现 `Color` 的 `Read` 实例的例子：

```

1  instance Read Color where
2  -- readsPrec is the main function for parsing input
3  readsPrec _ value =
4      -- We pass tryParse a list of pairs. Each pair has a string
5      -- and the desired return value. tryParse will try to match
6      -- the input to one of these strings
7      tryParse [("Red", Red), ("Green", Green), ("Blue", Blue)]
8      where
9          -- If there is nothing left to try, fail
10         tryParse [] = []
11         tryParse ((attempt, result) : xs) =
12             -- Compare the start of the string to be parsed to the
13             -- text we are looking for.
14             if take (length attempt) value == attempt
15             then -- If we have a match, return the result and the remaining input
16                 [(result, drop (length attempt) value)]
17             else -- If we don't have a match, try the next pair in the list of attempts.
18                 tryParse xs

```

测试：

```

1  ghci> (read "Red")::Color
2  Red
3  ghci> (read "Green")::Color
4  Green
5  ghci> (read "Blue")::Color
6  Blue
7  ghci> (read "[Red]")::[Color]
8  [Red]
9  ghci> (read "[Red,Red,Blue]")::[Color]
10 [Red,Red,Blue]
11 ghci> (read "[Red, Red, Blue]")::[Color]
12 *** Exception: Prelude.read: no parse

```

注意最后一个例子的异常。这是因为我们的解析器并没有聪明到能处理空格。

Tip

Read 并没有大范围的被使用

虽然可以使用 `Read` typeclass 来构建复杂的解析器，但许多人发现使用 `Parsec` 会更容易，且仅依赖 `Read` 来处理简单的任务。在第 16 章会详细介绍 `Parsec`。

通过 Read 和 Show 进行序列化

我们经常需要存储一个内存中的数据结构至硬盘供未来使用或者通过网络发送出去，那么将内存中数据转换为一个平坦的字节序列用于存储的这个过程就被称为序列化 *serialization*。

Tip

解析大型字符串

在 Haskell 中字符串的处理通常都是惰性的，因此 `read` 与 `show` 可以被用于处理很大的数据结构而不发生异常。Haskell 内建的 `read` 与 `show` 实例是高效的，并且都是纯 Haskell。如何处理解析异常的更多细节将会在第 19 章中详细介绍。

测试：

```
1 ghci> let d1 = [Just 5, Nothing, Nothing, Just 8, Just 9]::[Maybe Int]
2 ghci> putStrLn (show d1)
3 [Just 5,Nothing,Nothing,Just 8,Just 9]
4 ghci> writeFile "test" (show d1)
```

再是反序列：

```
1 ghci> input <- readFile "test"
2 "[Just 5,Nothing,Nothing,Just 8,Just 9]"
3 ghci> let d2 = read input
4
5 <interactive>:1:9:
6   Ambiguous type variable `a' in the constraint:
7     `Read a' arising from a use of `read' at <interactive>:1:9-18
8   Probable fix: add a type signature that fixes these type variable(s)
9 ghci> let d2 = (read input)::[Maybe Int]
10 ghci> print d1
11 [Just 5,Nothing,Nothing,Just 8,Just 9]
12 ghci> print d2
13 [Just 5,Nothing,Nothing,Just 8,Just 9]
14 ghci> d1 == d2
15 True
```

这里解释器并不知道 `d2` 的类型，因此抛出了异常。

以下是一些稍微复杂点的数据结构：

```
1 ghci> putStrLn $ show [( "hi", 1), ( "there", 3)]
2 [( "hi",1),( "there",3)]
3 ghci> putStrLn $ show [[1, 2, 3], [], [4, 0, 1], [], [503]]
4 [[1,2,3],[],[4,0,1],[],[503]]
5 ghci> putStrLn $ show [Left 5, Right "three", Left 0, Right "nine"]
6 [Left 5,Right "three",Left 0,Right "nine"]
7 ghci> putStrLn $ show [Left 0, Right [1, 2, 3], Left 5, Right []]
8 [Left 0,Right [1,2,3],Left 5,Right []]
```

数值类型

Haskell 拥有强大的数值类型。

选定的数值类型	
类型	描述
Double	双精度浮点数。浮点数的通常选择。
Float	单精度浮点数。通常用于与 C 交互。
Int	带方向的确定精度整数；最小范围 $[-2^{29}..2^{29} - 1]$ 。
Int8	8-bit 带方向的整数
Int16	16-bit 带方向的整数
Int32	32-bit 带方向的整数
Int64	64-bit 带方向的整数
Integer	带方向的任意精度整数；范围仅受机器限制。很常用。
Rational	带方向的任意精度的有理数。以两个 <code>Integers</code> 进行存储。
Word	无方向的确定精度整数；存储大小与 <code>Int</code> 一致
Word8	8-bit 无方向的整数
Word16	16-bit 无方向的整数
Word32	32-bit 无方向的整数
Word64	64-bit 无方向的整数

有很多不同的数值类型。有些运算，比如加法对所有类型都适用；还有其它类型的计算例如 `asin`，仅适用于浮点类型。

选定的数值函数与常量			
项	类型	模块	描述
(*)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	加法
(-)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	减法
(*)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	乘法
(/)	$\text{Fractional } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	除法
(**)	$\text{Floating } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	幂
()	$(\text{Num } a, \text{Integral } b) \Rightarrow a \rightarrow b \rightarrow a$	Prelude	非负数的幂
(i	$(\text{Fractional } a, \text{Integral } b) \Rightarrow a \rightarrow b$	Prelude	分数的幂
%)	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow \text{Ratio } a$	Data.Ratio	比例
(.&.)	$\text{Bits } a \Rightarrow a \rightarrow a \rightarrow a$	Data.Bits	Bitwise 和
(. .)	$\text{Bits } a \Rightarrow a \rightarrow a \rightarrow a$	Data.Bits	Bitwise 或
abs	$\text{Num } a \Rightarrow a \rightarrow a$	Prelude	绝对值
approxRational	$\text{RealFrac } a \Rightarrow a \rightarrow a \rightarrow \text{Rational}$	Data.Ratio	基于分数分子与分母的近似有理组合
cos	$\text{Floating } a \Rightarrow a \rightarrow a$	Prelude	Cosine, 还有 acos, cosh, 与 acosh
div	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数除法, 总是向下取整, quot 同理
fromInteger	$\text{Num } a \Rightarrow \text{Integer} \rightarrow a$	Prelude	从 Integer 类型转换为任意数值类型
fromIntegral	$(\text{Integral } a, \text{Num } b) \Rightarrow a \rightarrow b$	Prelude	比上述更泛化的转换
fromRational	$\text{Fractional } a \Rightarrow \text{Rational} \rightarrow a$	Prelude	从 Rational 转换, 可能有损
log	$\text{Floating } a \Rightarrow a \rightarrow a$	Prelude	自然 log
logBase	$\text{Floating } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	显式底的 log
maxBound	$\text{Bounded } a \Rightarrow a$	Prelude	bound 类型的最大值
minBound	$\text{Bounded } a \Rightarrow a$	Prelude	bound 类型的最小值
mod	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数模
pi	$\text{Floating } a \Rightarrow a$	Prelude	数学常数的派
quot	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数除法; 商的小数部分向零截断

接上表

项	类型	模块	描述
recip	Fractional a => a -> a	Prelude	倒数
rem	Integral a => a -> a -> a	Prelude	整数除法的余数
round	(RealFrac a, Integral b) => a -> b	Prelude	向最近的整数方向取整
shift	Bits a => a -> Int -> a	Bits	左移指定 bits, 右移可能为负
sin	Floating a => a -> a	Prelude	Sine, 另 asin, sinh, 与 asinh
sqrt	Floating a => a -> a	Prelude	平方根
tan	Floating a => a -> a	Prelude	Tangent, 另 atan, tanh, 与 atanh
toInteger	Integral a => a -> Integer	Prelude	转换任意 Integral 至一个 Integer
toRational	Real a => a -> Rational	Prelude	转换 Real 至 Rational
truncate	(RealFrac a, Integral b) => a -> b	Prelude	向零截断数值
xor	Bits a => a -> a -> a	Data.Bits	Bitwise 的异或

数值类型的 Typeclass 实例								
项	Bits	Bounded	Floating	Fractional	Integral	Num	Rea	RealFrac
Double			X	X		X	X	X
Float	X	X			X	X	X	
Int	X	X			X	X	X	
Int16	X	X			X	X	X	
Int32	X	X			X	X	X	
Int64	X	X			X	X	X	
Integer	X				X	X	X	
Rational or any Ratio				X		X	X	X
Word	X	X			X	X	X	
Word16	X	X			X	X	X	
Word32	X	X			X	X	X	
Word64	X	X			X	X	X	

在数值类型间转换是另一个常用的需求。

数值类型之间的转换				
原始类型	目标类型			
	Double, Float	Int, Word	Integer	Rational
Double, Float	fromRational . toRational	truncate *	truncate *	toRational
Int, Word	fromIntegral	fromIntegral	fromIntegral	fromIntegral
Integer	fromIntegral	fromIntegral	N/A	fromIntegral
Rational	fromRational	truncate *	truncate *	N/A

自动化派生

对于很多简单的数据类型,Haskell 编译器可以自动的为我们派生出 `Read` , `Show` , `Bounded` , `Enum` , `Eq` 以及 `Ord` 的实例。这将大大的节省用户编写代码的时间。

```
1 data Color = Red | Green | Blue
2 deriving (Read, Show, Eq, Ord)
```

测试:

```
1 ghci> show Red
2 "Red"
3 ghci> (read "Red")::Color
4 Red
5 ghci> (read "[Red,Red,Blue]")::[Color]
6 [Red,Red,Blue]
7 ghci> (read "[Red, Red, Blue]")::[Color]
8 [Red,Red,Blue]
9 ghci> Red == Red
10 True
11 ghci> Red == Blue
12 False
13 ghci> Data.List.sort [Blue,Green,Blue,Red]
14 [Red,Green,Blue,Blue]
15 ghci> Red < Blue
16 True
```

自动派生也不总是能成功。例如如果定义了一个类型 `data MyType = MyType (Int -> Bool)` ,编译器将不能为其派生 `show` 的实例,因为它不知道如何渲染一个函数。这种情况下我们会得到一个编译错误。

当我们自动派生某些 `typeclass` 的实例时,在数据声明中引用的类型也必须是该 `typeclass` 的实例(手动或自动的)。

```
1 data CannotShow = CannotShow
2 -- deriving (Show)
```

```

3
4  -- will not compile, since CannotShow is not an instance of Show
5  data CannotDeriveShow = CannotDeriveShow CannotShow
6      deriving (Show)
7
8  data OK = OK
9
10 instance Show OK where
11 show _ = "OK"
12
13 data ThisWorks = ThisWorks OK
14     deriving (Show)

```

注意 `CannotShow` 的 `deriving (Show)` 是被注释掉的，因此 `CannotDeriveShow` 才无法派生 `Show`。

Typeclasses 实战：令 JSON 使用起来更方便

上一章讲述的 `JValue` 处理 JSON 并不容易。例如，下面是一个实际 JSON 数据的删减整理的片段：

```

1  {
2    "query": "awkward squad haskell",
3    "estimatedCount": 3920,
4    "moreResults": true,
5    "results":
6    [{
7      "title": "Simon Peyton Jones: papers",
8      "snippet": "Tackling the awkward squad: monadic input/output ...",
9      "url": "http://research.microsoft.com/~simonpj/papers/marktoberdorf/",
10     },
11     {
12       "title": "Haskell for C Programmers | Lambda the Ultimate",
13       "snippet": "... the best job of all the tutorials I've read ...",
14       "url": "http://lambda-the-ultimate.org/node/724",
15     }
16   ]
17 }

```

以及在 Haskell 中的表达：

```

1  import SimpleJSON
2
3  result :: JValue
4  result = JObject [
5    ("query", JString "awkward squad haskell"),
6    ("estimatedCount", JNumber 3920),
7    ("moreResults", JBool True),
8    ("results", JArray [
9      JObject [

```

```

10     ("title", JString "Simon Peyton Jones: papers"),
11     ("snippet", JString "Tackling the awkward ..."),
12     ("url", JString "http://.../marktoberdorf/")
13 ])
14 ]

```

因为 Haskell 并不支持列表中包含不同类型的值，我们不能直接表示一个包含了多个类型的 JSON 对象。我们必须将每个值通过 `JValue` 构造函数来进行包装。这限制了我们的灵活性：如果想要更换数值 `3920` 成为一个字符串 `"3,920"`，我们必须更换构造函数，即 `JNumber` 变为 `JString`。

Haskell 的 typeclasses 为此类问题提供了一个诱人的解决方案：

```

1  type JSONError = String
2
3  class JSON a where
4      toJValue :: a -> JValue
5      fromJValue :: JValue -> Either JSONError a
6
7  instance JSON JValue where
8      toJValue = id
9      fromJValue = Right

```

现在无需再将类似 `JNumber` 这样的构造函数应用在值上将值包裹，直接应用 `toJValue` 函数在该值上即可。

我们同样提供了一个 `fromJValue` 函数，用于将一个 `JValue` 转换成一个我们所期望的类型。

更多有帮助的错误

让我们构造一下自己的 `Maybe` 与 `Either`：

```

1  data Maybe a
2  = Nothing
3  | Just a
4  deriving (Eq, Ord, Read, Show)
5
6  data Either a b
7  = Left a
8  | Right b
9  deriving (Eq, Ord, Read, Show)

```

在 `Bool` 值实例中尝试一下：

```

1  instance JSON Bool where
2      toJValue = JBool
3      fromJValue (JBool b) = Right b
4      fromJValue _ = Left "not a JSON boolean"

```

通过类型同义词来创建一个实例

```

1 instance JSON String where
2   toJValue = JString
3   fromJValue (JString s) = Right s
4   fromJValue _ = Left "not a JSON string"

```

活在开放世界

Haskell 的 typeclasses 允许我们在任何合适的时候创建新的 typeclass 实例。

```

1 doubleToJValue :: (Double -> a) -> JValue -> Either JSONError a
2 doubleToJValue f (JNumber v) = Right (f v)
3 doubleToJValue _ _ = Left "not a JSON number"
4
5 instance JSON Int where
6   toJValue = JNumber . realToFrac
7   fromJValue = doubleToJValue round
8
9 instance JSON Integer where
10  toJValue = JNumber . realToFrac
11  fromJValue = doubleToJValue round
12
13 instance JSON Double where
14  toJValue = JNumber
15  fromJValue = doubleToJValue id

```

我们还希望转换一个列表成为 JSON，暂时用 `undefined` 作为实例的方法。

```

1 instance (JSON a) => JSON [a] where
2   toJValue = undefined
3   fromJValue = undefined

```

对象亦是如此：

```

1 instance (JSON a) => JSON [(String, a)] where
2   toJValue = undefined
3   fromJValue = undefined

```

合适重叠实例会导致问题

如果我们将这些定义放入一个原文件中，并加载至 `ghci`，每个初始化看起来都没问题：

```

1 ghci> :load BrokenClass
2 [1 of 2] Compiling SimpleJSON      ( ../ch05/SimpleJSON.hs, interpreted )
3 [2 of 2] Compiling BrokenClass      ( BrokenClass.hs, interpreted )
4 Ok, modules loaded: SimpleJSON, BrokenClass.

```

然而当我们尝试使用元组的列表时，错误便发生了。

```

1 ghci> toJValue [("foo", "bar")]
2
3 <interactive>:1:0:
4   Overlapping instances for JSON [(Char), (Char)]
5     arising from a use of `toJValue' at <interactive>:1:0-23
6   Matching instances:
7     instance (JSON a) => JSON [a]
8       -- Defined at BrokenClass.hs:(44,0)-(46,25)
9     instance (JSON a) => JSON [(String, a)]
10      -- Defined at BrokenClass.hs:(50,0)-(52,25)
11   In the expression: toJValue [("foo", "bar")]
12   In the definition of `it': it = toJValue [("foo", "bar")]

```

重叠实例的问题是 Haskell 的开放世界假设的结果。下面用一个更简单的例子来说明到底发生了什么：

```

1 class Borked a where
2   bork :: a -> String
3
4 instance Borked Int where
5   bork = show
6
7 instance Borked (Int, Int) where
8   bork (a, b) = bork a ++ ", " ++ bork b
9
10 instance (Borked a, Borked b) => Borked (a, b) where
11   bork (a, b) = ">>" ++ bork a ++ " " ++ bork b ++ "<<"

```

我们有两个 typeclass `Borked` 的 pairs 实例：一对是 `Int`，另一对是其它任意值。

假设我们希望 `bork` 一对 `Int` 值，那么编译器必须选择一个实例来使用。由于这些实例相邻，那么看起来就能够简单的选择更加明确的实例。

然而 GHC 默认是保守的，它会坚持必须只有一个可能的实例。因此最终在使用 `bork` 时会抛出异常。

Tip

什么时候重叠的实例会有影响？

正如我们之前提到的那样，我们可以将一个 typeclass 的实例散落在若干模块中。GHC 并不会抱怨重叠实例的存在。而真正抱怨的时候就是当我们尝试使用这个受影响 typeclass 的方法时，也就是当强制要求选择哪一个实例需要使用的時候。

放宽 typeclasses 的某些限制

通常而言，我们不可以编写一个特定多态类型的 typeclass 实例。例如 `[Char]` 类型就是 `[a]` 指定 `Char` 类型的多态。因此禁止将 `[Char]` 声明为 typeclass 的实例。这个非常的不

方便，因为字符串在真实代码中处处存在。

`TypeSynonymInstances` 语言扩展移除了这个限制，允许我们编写上述的实例。

GHC 还支持另一个有用的语言扩展，`OverlappingInstances`，专门用于处理重叠实例。当存在若干重叠实例需要选择时，该扩展使编译器选择最明确的那个。

我们通常将该扩展与 `TypeSynonymInstances` 一起使用。例如：

```
1 import Data.List
2
3 class Foo a where
4     foo :: a -> String
5
6 instance {-# OVERLAPS #-} Foo a => Foo [a] where
7     -- foo = concat . intersperse ", " . map foo
8     foo = intercalate ", " . map foo
9
10 instance {-# OVERLAPS #-} Foo Char where
11     foo c = [c]
12
13 instance {-# OVERLAPS #-} Foo String where
14     foo = id
15
16 main :: IO ()
17 main = do
18     putStrLn $ "foo: " ++ foo "SimpleClass"
19     putStrLn $ "foo: " ++ foo ["a", "b", "c"]
```

与原文不同之处在于，注解 `{-# LANGUAGE OverlappingInstances #-}` 在 6.8.1 后被弃用，现在则是在 `instance` 后使用 `{-# OVERLAPS #-}` 来表示重叠的实例，详见文档。

如果我们将 `foo` 应用至一个 `String`，编译器将使用 `String` 指定的实现。尽管存在 `[a]` 与 `Char` 的 `Foo` 实例，但是 `String` 的实例更加的明确，因此 GHC 会选择它。

如何给类型一个新的身份

略。

```
1 data DataInt = D Int
2     deriving (Eq, Ord, Show)
3
4 newtype NewtypeInt = N Int
5     deriving (Eq, Ord, Show)
```

略。

总结：三总命名类型的方法：

- `data` 关键字引入了一个真实的新的代数数据类型。

- `type` 关键字给予了已存在的类型一个同义词。我们可以替换的使用类型与其同义词。
- `newtype` 关键字给予已存在类型一个新的身份。原始类型和新的类型是不可替换的。

JSON typeclasses 没有重叠的实例

现在需要帮助编译器区分 JSON 数组的 `[a]`，以及 JSON 对象的 `[(String, [a])]`。它们是造成重叠问题的原因。我们将列表类型包裹起来，这样编译器则不会视其为列表：

```
1 newtype JAry a = JAry {fromJAry :: [a]}
2 deriving (Eq, Ord, Show)
```

当需要将其从模块中导出时，我们需要导出该类型的所有细节。我们模块头看起来像是这样：

```
1 module JSONClass (JAry (...)) where
```

接下来是另一个包装类型用于隐藏 JSON 对象：

```
1 newtype JObj a = JObj {fromJObj :: [(String, a)]}
2 deriving (Eq, Ord, Show)
```

有了这些类型定义，那么就可以对 `JValue` 类型做点小修改：

```
1 data JValue
2   = JString String
3   | JNumber Double
4   | JBool Bool
5   | JNull
6   | JObject (JObj JValue) -- was [(String, JValue)]
7   | JArray (JAry JValue) -- was [JValue]
8   deriving (Eq, Ord, Show)
```

这个改动并不会影响已经写过的 `JSON` typeclass 实例，不过还是需要为 `JAry` 与 `JObj` 类型编写 `JSON` typeclass 实例。

```
1 jaryFromJValue :: (JSON a) => JValue -> Either JSONError (JAry a)
2
3 jaryToJValue :: (JSON a) => JAry a -> JValue
4
5 instance (JSON a) => JSON (JAry a) where
6   toJValue = jaryToJValue
7   fromJValue = jaryFromJValue
```

让我们慢慢的看一下将 `JAry a` 转换成 `JValue` 的每个步骤。给定一个列表，我们知道其所有元素都是 `JSON` 实例，将其转换成一个列表的 `JValue` 很简单。

```
1 listToJValues :: (JSON a) => [a] -> [JValue]
2 listToJValues = map toJValue
```

有了上述代码后，将其变为一个 `JArray JValue` 就仅仅是应用 `newtype` 类型构造函数即可：

```
1 jvaluesToJArray :: [JValue] -> JArray JValue
2 jvaluesToJArray = JArray
```

（记住这并没有性能损耗，仅仅只是告诉编译器隐藏我们在使用一个列表的事实。）将其转换成一个 `JValue`，我们应用赢一个类型构造函数：

```
1 jarrayOfJValuesToJValue :: JArray JValue -> JValue
2 jarrayOfJValuesToJValue = JArray
```

将这些部分用函数组合的方式集成起来，就获得了一个简洁的一行代码转换成一个 `JValue`：

```
1 jarrayToJValue = JArray . JArray . map toJValue . fromJArray
```

从 `JValue` 转换至一个 `JArray a` 则需要更多的工作，我们还是将其分解成可复用的部分。基础函数很直接：

```
1 jarrayFromJValue (JArray (JArray a)) = whenRight JArray (mapEithers fromJValue a)
2 jarrayFromJValue _ = Left "not a JSON array"
```

`whenRight` 函数检查它的参数：如果是由 `Right` 构造函数构建的它则调用一个函数，如果是由 `Left` 构造则保留值不变：

```
1 whenRight :: (b -> c) -> Either a b -> Either a c
2 whenRight _ (Left err) = Left err
3 whenRight f (Right a) = Right (f a)
```

更复杂的是 `mapEithers`，它的行为类似于普通的 `map` 函数，但是如果遇到一个 `Left` 值，它会立刻返回，而不是继续累积一个 `Right` 值的列表。

```
1 mapEithers :: (a -> Either b c) -> [a] -> Either b [c]
2 mapEithers f (x : xs) = case mapEithers f xs of
3   Left err -> Left err
4   Right ys -> case f x of
5     Left err -> Left err
6     Right y -> Right (y : ys)
7 mapEithers _ _ = Right []
```

由于隐藏在 `JObj` 类型的列表中的元素有少许结构，因此它与 `JValue` 之间的转换会变得更复杂。幸运的是我们可以复用刚刚定义好的函数：

```
1 instance (JSON a) => JSON (JObj a) where
2   toJValue = JObject . JObj . map (second toJValue) . fromJObj
3   fromJValue (JObject (JObj o)) = whenRight JObj (mapEithers unwrap o)
4   where
5     unwrap (k, v) = whenRight (k,) (fromJValue v)
6   fromJValue _ = Left "not a JSON object"
```

7 Input and output

Haskell 中的经典 I/O

略

处理文件以及句柄

Haskell 为 I/O 定义了相当多的基本函数，其中许多函数与其它编程语言中的函数相似。`System.IO` 库中提供了所有的基础 I/O 函数。

使用 `openFile` 会返回一个文件的 `Handle`，它用于对文件执行指定的操作。Haskell 提供了例如 `hPutStrLn` 这样的函数，其类似于 `putStrLn`，不同在于接受一个额外的参数 – 一个 `Handle` – 指定哪个文件被操作。当我们结束时，需要用 `hClose` 来结束 `Handle`。这些函数都定义在 `System.IO` 中，“h”开头的函数对应了几乎所有的非“h”开头的函数；例如 `print` 打印在屏幕上，而 `hPrint` 打印至一个文件。

一个例子：

```

1  import Data.Char (toUpper)
2  import System.IO
3
4  main :: IO ()
5  main = do
6      inh <- openFile "input.txt" ReadMode
7      outh <- openFile "output.txt" WriteMode
8      mainloop inh outh
9      hClose inh
10     hClose outh
11
12     putStrLn "whatever"
13
14 mainloop :: Handle -> Handle -> IO ()
15 mainloop inh outh = do
16     ineof <- hIsEOF inh
17     if ineof
18     then return ()
19     else do
20         inpStr <- hGetLine inh
21         hPutStrLn outh (map toUpper inpStr)
22         mainloop inh outh

```

`mainloop` 首先检查输入是否结束 (EOF)，如果没有则读取一行，将该行转为大写后写入输出文件中，再递归的调用 `mainloop`。

略

可能的 IOMode 值				
IOMode	可读?	可写?	起始位置	说明
ReadMode	Yes	No	文件起始	文件必须存在
WriteMode	No	Yes	文件起始	文件存在时会被清空
ReadWriteMode	Yes	Yes	文件起始	文件不存在时创建；否则现有数据保留
AppendMode	No	Yes	文件尾部	文件不存在时创建；否则现有数据保留

关闭句柄

上述例子中我们已经见到了 `hClose` 用于关闭文件的句柄。在之后的小节中我们将了解缓存 Buffering 这个概念，即 Haskell 在内部为文件维护了缓存。这样做能显著的提升性能，然而这就意味着在对打开的文件调用 `hClose` 之前，数据可能不会被刷新到操作系统中。

另一个原因是打开着的文件消耗系统的资源。如果程序运行的时间很长，开开了很多文件但是并没有关闭它们，那么程序很可能会因为资源枯竭而导致崩溃。

Seek 与 Tell

从一个关联了磁盘文件的 `Handle` 读写时，操作系统会维护一个关于当前位置的内部记录。每当进行下一层读取的时，操作系统返回下一个数据块，其起始点为当前位置，且增加的位置反应了读取了多少数据。

你可以使用 `hTell` 来找到文件中当前的位置。当文件被创建时，它是空的且你的位置将会是 0。在编写 5 个字节后，你的位置则变为了 5，以此类推。`hTell` 接受一个 `Handle` 并返回一个 `IO Integer` 来表示你的位置。

`hTell` 的同伴是 `hSeek`，它允许你更改文件的位置，其三个参数为：`Handle`，`SeekMode` 以及一个位置。

`SeekMode` 有三个不同类型，用于指定如何解析给定的位置。`AbsoluteSeek` 为文件中的精确位置，等同于 `hTell`；`RelativeSeek` 意为当前位置开始的多少位置，正值向后，负值向前；`SeekFromEnd` 则是从文件末尾往前多少的位置。

并不是所有的 `Handle` 都是可以 seek 的。一个 `Handle` 通常关联一个文件，但是它还可以关联其它的东西，比如网络连接，磁带机，或者终端。可以使用 `hIsSeekable` 来查看给定的 `Handle` 是否可以 seek。

标准输入，输出以及错误

较早之前我们指出每个非“h”函数通常都会有与其关联的“h”函数用作于任何 `Handle` 上。实际上非“h”函数只不过是“h”函数的一种缩写。

在 `System.IO` 中有三个预定义的 `Handle` : `stdin` 标准输入, 通常是键盘; `stdout` 标准输出, 通常是显示器; `stderr` 标准错误, 通常也是显示器。

像是 `getLine` 类似的函数可以简单的定义成:

```
1 getLine = hGetLine stdin
2 putStrLn = hPutStrLn stdout
3 print = hPrint stdout
```

删除与文件重命名

`System.Directory` 模块提供了两个比较有用的函数: `removeFile` 接受单个参数, 即文件名, 然后删除该文件; `renameFile` 接受两个文件名, 一个旧名称以及一个新的名称, 如果新的文件名是一个不同的路径, 那么可以认为这是一个移动。旧的文件名必须在调用 `renameFile` 之前就存在, 另外如果新文件已经存在, 则重命名后将其删除。

跟很多其它接受一个文件名的函数一样, 如果“旧”名称不存在, `renameFile` 则会抛出异常。

`System.Directory` 模块中还有很多其他的函数, 像是创建或移除文件夹, 在路径中查找文件列表, 测试文件是否存在, 等等。

临时文件

程序员频繁的需要临时文件。这些文件被用于存储大量等待计算的数据, 可被其它程序所用的数据, 等等。

通过名为 `openTempFile` 的函数 (以及相关联的 `openBinaryTempFile`) 可以帮助我们解决不少问题。

`openTempFile` 接受两个参数: 创建文件的路径, 以及一个“template”用于命名文件。路径可以是 `.` 来代表当前路径, 或者也可以使用 `System.Directory.getTemporaryDirectory` 来找到操作系统所给出的最佳放置临时文件的位置。template 则将一些随机字符添加至文件, 用以确保结果是真正唯一的。实际上它保证了可以在一个唯一的文件名上工作。

`openTempFile` 的返回类型是 `IO (FilePath, Handle)` 。元组的第一部分是被创建文件的名称, 第二部分则是一个模式为 `ReadWriteMode` 的打开了文件的 `Handle` 。当我们使用完文件, 我们希望用 `hClose` 来操作 `Handle` 用于关闭文件, 接着调用 `removeFile` 来删除它。接下来我们会看到一个例子用于展示这些函数的用法。

扩展案例: 函数式 I/O 以及临时文件

```
1 import Control.Exception
2 import System.Directory (getTemporaryDirectory, removeFile)
3 import System.IO
```

```

4
5 main :: IO ()
6 main = withTempFile "mytemp.txt" myAction
7
8 myAction :: FilePath -> Handle -> IO ()
9 myAction tempname temp = do
10     -- Start by displaying a greeting on the terminal
11     putStrLn "Welcome to tempfile.hs"
12     putStrLn $ "I have a temporary file at " ++ tempname
13
14     -- Let's see what the initial position is
15     pos <- hTell temp
16     putStrLn $ "My initial position is " ++ show pos
17
18     -- Now, write some data to the temporary file
19     let tempdata = show [1 .. 10]
20     putStrLn $ "Writing one line containing " ++ show (length tempdata) ++ " bytes: " ++
        tempdata
21     hPutStrLn temp tempdata
22
23     -- Get our new position. This doesn't actually modify pos in memory,
24     -- but makes the name "pos" correspond to a different value for
25     -- the remainder of the "do" block.
26     pos <- hTell temp
27     putStrLn $ "After writing, my new position is " ++ show pos
28
29     -- Seek to the beginning of the file and display it
30     putStrLn "The file content is: "
31     hSeek temp AbsoluteSeek 0
32
33     -- hGetContents performs a lazy read of the entire file
34     c <- hGetContents temp
35
36     -- Copy the file byte-for-byte to stdout, followed by \n
37     putStrLn $ "c: " ++ c
38
39     -- Let's also display it as a Haskell literal
40     putStrLn "Which could be expressed as this Haskell literal:"
41     print c
42
43 {-
44     This function takes two parameters: a filename pattern and another function.
45     It will create a temporary file, and pass the name and Handle of that file to
46     the given function.
47
48     The temporary file is created with openTempFile. The directory is the one
49     indicated by getTemporaryDirectory, or, if the system has no notion of a temporary
50     directory, "." is used. The given pattern is passed to openTempFile.
51

```

```

52     After the given function terminates, even if it terminates due to an exception,
53     the Handle is closed and the file is deleted.
54 -}
55 withTempFile :: String -> (FilePath -> Handle -> IO a) -> IO a
56 withTempFile pattern func = do
57     -- The library ref says that getTemporaryDirectory may raise an exception on
58     -- systems that have no notion of a temporary directory. So, we run
59     -- getTemporaryDirectory under catch. catch takes two functions: one to run,
60     -- and a different one to run if the first raised an exception.
61     -- If getTemporaryDirectory raised an exception, just use "." (the current
62     -- working directory).
63     tempdir <-
64         catch
65             getTemporaryDirectory
66             (\(_ :: IOException) -> return ".") -- explicit annotates exception type
67     (tempfile, temp) <- openTempFile tempdir pattern
68
69     -- Call (func tempfile temp) to perform the action on the temporary file.
70     -- finally takes two actions. The first is the action to run, the second is an action
71     -- to run after the first, regardless of the temporary file is always deleted.
72     -- The return value from finally is the first action's return value.
73     finally
74         (func tempfile temp)
75         ( do
76             hClose temp
77             removeFile tempfile
78         )

```

首先 `withTempFile` 函数证明了 Haskell 在 I/O 时也没有忘记其函数式的天性。该函数接受一个 `String` 与另一个函数。传递给 `withTempFile` 的函数带着临时文件的 `Handle` 被调用，当该函数退出时，临时文件关闭并被删除。

异常处理可以让程序变得更加健壮。通常我们希望在程序结束后删除临时文件，即使程序的执行过程中出现了错误。更多的错误处理会在第十九章讲述。

现在回到程序，`main` 做的事情很简单，调用函数 `withTempFile` 并传入临时文件名以及 `myAction` 函数。

`myAction` 展示了一些信息到终端上，写入了一些数据到文件中，`seek` 到文件的起始位置，通过 `hGetContents` 进行数据读取，接着按照字节展示文件内容，同时通过 `print c` 进行打印。这等同于 `putStrLn (show c)`。

现在看看输出：

```

1 % runhaskell tempfiles.hs
2 Welcome to tempfile.hs
3 I have a temporary file at /var/folders/nb/w1q3ztlj139_vz9pqglmbks80000gn/T/mytemp3709-0.txt
4 My initial position is 0
5 Writing one line containing 22 bytes: [1,2,3,4,5,6,7,8,9,10]
6 After writing, my new position is 23

```

```

7 The file content is:
8 c: [1,2,3,4,5,6,7,8,9,10]
9
10 Which could be expressed as this Haskell literal:
11 "[1,2,3,4,5,6,7,8,9,10]\n"

```

Lazy I/O

Haskell 还提供了另一种方式处理 I/O。由于 Haskell 是 lazy 语言，意味着任何数据只会在其值必须被知道时才会被计算，那么就有了一些新颖的 I/O 处理方法。

hGetContents

一个新颖的 I/O 方式就是 `hGetContents` 函数，其类型 `Handle -> IO String`，这里的 `String` 代表着由文件的 `Handle` 所返回的所有数据。

在严格求值的语言中，使用这样的函数通常是一个坏主意。假设读一个 500GB 的文件，那么就有可能因为内存不足而导致程序崩溃。在这些语言中，传统的做法就是使用循环来处理文件的所有数据。

但是 `hGetContents` 不一样，它返回的 `String` 是惰性的，即在调用 `hGetContents` 时，并没有读取任何数据。只有在处理列表的元素（字符）时才会从 `Handle` 中读取数据。当 `String` 的元素不再使用时，Haskell 的垃圾收集器则会自动释放内存。

来看一个例子：

```

1 import Data.Char (toUpper)
2 import System.IO
3
4 main :: IO ()
5 main = do
6     inh <- openFile "input.txt" ReadMode
7     outh <- openFile "output.txt" WriteMode
8     inpStr <- hGetContents inh
9     let result = processData inpStr
10    hPutStr outh result
11    hClose inh
12    hClose outh
13
14 processData :: String -> String
15 processData = map toUpper

```

测试：

```

1 ghci> :load toupper-lazy1.hs
2 [1 of 1] Compiling Main           ( toupper-lazy1.hs, interpreted )
3 Ok, modules loaded: Main.
4 ghci> processData "Hello, there! How are you?"

```



```
5 "HELLO, THERE! HOW ARE YOU?"
6 ghci> :type processData
7 processData :: String -> String
8 ghci> :type processData "Hello!"
9 processData "Hello!" :: String
```

Warning

如果我们在上述例子中尝试在使用 `inpStr` 的地方（对 `processData` 的调用）之后继续使用 `inpStr`，那么程序则会失去其内存效率。这是因为编译器将会强制保留 `inpStr` 值在内存中供未来使用。在这里编译器知道 `inpStr` 不会再被使用，那么就在其使用结束后尽快的释放内存。这里需要记住：内存只有在最后一次使用后才会被释放。

这段代码有点啰嗦，我们可以让它变得更简洁一些：

```
1 import Data.Char (toUpper)
2 import System.IO
3
4 main :: IO ()
5 main = do
6   inh <- openFile "input.txt" ReadMode
7   outh <- openFile "output.txt" WriteMode
8   inpStr <- hGetContents inh
9   hPutStr outh (map toUpper inpStr)
10  hClose inh
11  hClose outh
```

在使用 `hGetContents` 时，我们甚至不需要从输入文件中消费所有的数据。每当 Haskell 系统确定 `hGetContents` 所返回的整个字符可以被垃圾回收时 – 意味着它不会再被使用 – 文件将自动被关闭。同样的原则也适用于从文件中读取的数据。每当给定的数据不再需要时，Haskell 环境就会释放存储该数据的内存。严格来说我们完全都不需要调用 `hClose`。然而这是一个好的习惯，因为之后对程序的改动会使得 `hClose` 变得很重要。

Warning

在使用 `hGetContents` 时，必须记住即使在之后的程序可能不再显式的引用 `Handle`，但还是要保持 `Handle` 不被关闭，直到 `hGetContents` 的结果被消费掉了。不这么做的话就可能会导致丢失部分或者全部文件数据。这是因为 Haskell 是惰性的，所以通常可以假设只有在输出了涉及输入的计算结果之后才消费了输入。

readFile 与 writeFile

Haskell 程序员经常使用 `hGetContents` 作为筛选。从一个文件读取数据，处理数据，接着将数据写到别的地方。这非常的通用所以有了一些快捷的方式：`readFile` 以及 `writeFile` 就是这样的函数，以字符串的方式处理文件。它们处理了所有关于打开文件，关闭文件，读取数据，以及写入数据的细节。`readFile` 在内部使用了 `hGetContents`。

通过 `ghci` 查看这些函数的类型：

```
1 ghci> :type readFile
2 readFile :: FilePath -> IO String
3 ghci> :type writeFile
4 writeFile :: FilePath -> String -> IO ()
```

使用 `readFile` 与 `writeFile` 改造之前的代码：

```
1 import Data.Char (toUpper)
2
3 main :: IO ()
4 main = do
5     inpStr <- readFile "input.txt"
6     writeFile "output.txt" $ map toUpper inpStr
```

关于 Lazy Output

现在我们能理解惰性输入是如何在 Haskell 中工作的。那么惰性输出又是怎么样呢？

我们知道 Haskell 中所有计算都是在其值被需要时才会进行。由于函数例如 `writeFile` 以及 `putStr` 会输出所有传给它们的 `String`，那么整个 `String` 必须被计算。因此可以保证 `putStr` 的参数将被完整求值。

那么输入惰性的意义呢？上述例子中，对 `putStr` 或 `writeFile` 的调用是否会强制将整个输入字符串立刻加载到内存中，仅仅只是为了输出？

回答是不。`putStr`（以及其它类似的所有输出函数）在数据可同时，输出数据。它们同样也不需要保留已经输出了的数据，因此只要程序中没有其它东西需要它，内存就可以立刻释放。在某种意义上，可以将 `readFile` 与 `writeFile` 之间的 `String` 视为连接两者的管道。数据进入一端，以某种方式转换后，从另一端流出。

interact

通过 `interact` 函数可以进一步简化我们的程序：

```
1 import Data.Char (toUpper)
2
3 main :: IO ()
4 main = interact $ map toUpper
```

仅用了一行！测试：

```
1 $ runghc toupper-lazy4.hs < input.txt > output.txt
```

或者想要打印到屏幕上：

```
1 $ runghc toupper-lazy4.hs < input.txt
```

如果想要看到 Haskell 的输出确实是在接收到数据块后立刻输出到数据块，可以不加任何参数运行 `runghc toupper-lazy4.hs`，这样就可以看到在敲击输入回车后，所有字符都会被立刻以大写的形式输出。

我们同样还可以通过 `interact` 编写在输出前添加新行的程序：

```
1 import Data.Char (toUpper)
2
3 main = interact $ map toUpper . (++) "Your data, in uppercase, is:\n\n"
```

由于我们在 `(++)` 后调用的 `map`，那么新行也会变为大写，修改一下：

```
1 import Data.Char (toUpper)
2
3 main = interact $ (++) "Your data, in uppercase, is:\n\n" . map toUpper
```

这就将新行移到 `map` 外去了。

interact 加上过滤

`interact` 的另一个常用方法就是过滤。比如打印带有字符“a”的每行：

```
1 main = interact $ unlines . filter (elem 'a') . lines
```

通过 `ghci` 查看上述的三个新函数：

```
1 ghci> :type lines
2 lines :: String -> [String]
3 ghci> :type unlines
4 unlines :: [String] -> String
5 ghci> :type elem
6 elem :: (Eq a) => a -> [a] -> Bool
```

测试：

```
1 $ runghc filter.hs < input.txt
2 I like Haskell
3 Haskell is great
```

IO 单子

Actions

大多数语言不会区分纯函数和非纯函数。Haskell 的函数有其数学意义：它们是存粹的计算，且不会被任何外界的东西改变。另外，计算可以在任何时刻进行。

那么就需要另一些工具来处理 I/O 了。这个工具在 Haskell 中叫做 *actions*。Actions 类似于函数，在定义的时候不会做任何事情，而是在执行任务时才会被唤醒。I/O actions 被定义在 **IO** 单子中。我们会在第十四章中讲解单子。看看一些类型：

```
1 ghci> :type putStrLn
2 putStrLn :: String -> IO ()
3 ghci> :type getLine
4 getLine :: IO String
```

`putStrLn` 的类型跟其它函数无异。函数接受一个参数并返回一个 `IO ()`。这个 `IO ()` 就是 action。我们可以存储以及传递 actions 至纯代码中，尽管这并不常用。一个 action 不会做任何事情直到它被唤起。看看例子：

```
1 str2action :: String -> IO ()
2 str2action input = putStrLn $ "Data: " ++ input
3
4 list2actions :: [String] -> [IO ()]
5 list2actions = map str2action
6
7 numbers :: [Int]
8 numbers = [1 .. 10]
9
10 strings :: [String]
11 strings = map show numbers
12
13 actions :: [IO ()]
14 actions = list2actions strings
15
16 printitall :: IO ()
17 printitall = runall actions
18
19 -- Take a list of actions, and execute each of them in turn.
20 runall :: [IO ()] -> IO ()
21 runall [] = return ()
22 runall (fst : remaining) = do
23     fst
24     runall remaining
25
26 main :: IO ()
27 main = do
28     str2action "Start of the program"
29     printitall
```

```
30 str2action "Done!"
```

`str2action` 这个函数接受一个参数并返回一个 `IO ()`。在 `main` 的结尾，可以在另一个 action 中直接使用它，并打印出一行。或者可以存储 – 而不是执行 – 这个 action 到纯代码中。在 `list2actions` 中 – 使用了 `map` 应用在 `str2action` 上并返回一个列表的 actions，正如我们在纯代码中的那样。可以看到通过 `printitall` 的所有内容都是使用纯的工具构建的。

尽管定义了 `printitall`，它不会被执行直到它的 action 在某处被计算。注意在 `main` 中是如何将 `str2action` 作为一个 I/O action 被执行的，而之前我们在 I/O 单子外使用它，并将结果转为一个列表。

可以这么理解：每个声明，除了 `let`，在一个 `do` 代码块中必须产生一个将要执行的 I/O action。

对 `printitall` 的调用最终执行所有这些操作。实际上由于 Haskell 是惰性的，所以直到这里才会生成 actions。

测试：

```
1 % runhaskell actions.hs
2 Data: Start of the program
3 Data: 1
4 Data: 2
5 Data: 3
6 Data: 4
7 Data: 5
8 Data: 6
9 Data: 7
10 Data: 8
11 Data: 9
12 Data: 10
13 Data: Done!
```

实际上我们可以将其写成更紧凑的形式。以下是重构：

```
1 str2message :: String -> String
2 str2message input = "Data: " ++ input
3
4 str2action :: String -> IO ()
5 str2action = putStrLn . str2message
6
7 numbers :: [Int]
8 numbers = [1 .. 10]
9
10 main :: IO ()
11 main = do
12     str2action "Start of the program"
13     mapM_ (str2action . show) numbers
14     str2action "Done!"
```

注意 `str2action` 中使用了标准函数组合操作符。在 `main` 中，调用了 `mapM_`，该函数类似于 `map` 接受一个函数与一个列表。而提供给 `mapM_` 的函数则是一个应用至列表中每个元素的 I/O action。`mapM_` 会抛出函数的结果，不过如果有需要，也可以使用 `mapM` 返回 I/O 结果的列表。看一下它们的类型：

```
1 ghci> :type mapM
2 mapM :: (Monad m) => (a -> m b) -> [a] -> m [b]
3 ghci> :type mapM_
4 mapM_ :: (Monad m) => (a -> m b) -> [a] -> m ()
```

Tip

实际上这些函数并不仅仅作用于 I/O；它们可用于任何 `Monad`。

为什么有了 `map` 还需要 `mapM` 呢？因为 `map` 是一个返回列表的纯函数，它不能直接执行 actions，而 `mapM` 则是 IO 单子中的工具，可用于实际执行 actions。

回到 `main`，`mapM_` 应用了 `(str2action . show)` 在每个 `numbers` 中的元素上。`show` 将每个数值转为 `String`，而 `str2action` 转换每个 `String` 为一个 action。`mapM_` 结合这些独立的 actions 成为一个大的 action 并打印出来。

Sequencing

`do` 代码块其实是合并所有 actions 的简便注解。在不使用 `do` 时还有两个操作符可以使用：`>>` 与 `>>=`。

```
1 ghci> :type (>>)
2 (>>) :: (Monad m) => m a -> m b -> m b
3 ghci> :type (>>=)
4 (>>=) :: (Monad m) => m a -> (a -> m b) -> m b
```

`>>` 操作符将两个 actions 串联起来：先执行第一个 action，再执行第二个 action。整个计算的结果则是第二个 action 的结果，而第一个 action 的结果则被丢弃。这类似于 `do` 代码块中的一行。可以用 `putStrLn "line 1" >> putStrLn "line 2"` 来测试一下。它将会打印出两行，将第一个 `putStrLn` 的结果丢弃，将第二个的结果作为返回。

`>>=` 操作符运行一个 action，然后将其结果传给一个返回 action 的函数。第二个 action 也将运行，整个表达式的结果还是第二个 action 的结果。例如，可以用 `getLine >>= putStrLn` 测试，它从键盘读取一行然后再将其输出。

现在不用 `do` 代码块来重写一下之前的例子：

```
1 main :: IO ()
2 main = do
3     putStrLn "Greetings! What is your name?"
```

```

4 inpStr <- getLine
5 putStrLn $ "Welcome to Haskell, " ++ inpStr ++ "!"

```

不使用 `do` :

```

1 main :: IO ()
2 main =
3   putStrLn "Greetings! What is your name?"
4   >> getLine
5   >>= (\inpStr -> putStrLn $ "Welcome to Haskell, " ++ inpStr ++ "!")

```

在定义 `do` 代码块时, Haskell 编译器内部会执行类似的转换。

Return 的本质

在本章开头我们提到了 `return` 可能并不像是它长得那样。很多语言的关键字 `return` 是立刻终结函数的执行并返回一个值给到调用者。

Haskell 中的 `return` 函数则大相径庭。Haskell 中的 `return` 将数据包装进单子中。在谈及 I/O 时, `return` 用于将纯数据放入 IO 单子中。

为什么要这么做呢? 记住, 任何结果依赖于 I/O 的东西都必须在 IO 单子内。因此, 如果我们正在编写一个执行 I/O 的函数, 那么一个纯计算, 我们需要使用 `return` 来使这个纯计算成为函数的正确返回值。例如:

```

1 import Data.Char (toUpper)
2
3 isGreen :: IO Bool
4 isGreen = do
5   putStrLn "Is green your favorite color?"
6   inpStr <- getLine
7   return $ (toUpper . head $ inpStr) == 'Y'

```

我们有一个返回 `Bool` 的纯计算。计算传递给 `return`, 将其放入 IO 单子内。由于它是 `do` 代码块中的最后一个值, 它变成了 `isGreen` 的返回值, 但这并不因为我们使用了 `return` 函数。

下面是同样一个程序, 只不过将纯计算剥离成为一个独立的函数。这样可以隔离纯函数, 同样使得意图更加的清晰。

```

1 import Data.Char (toUpper)
2
3 isYes :: String -> Bool
4 isYes inpStr = (toUpper . head $ inpStr) == 'Y'
5
6 isGreen :: IO Bool
7 isGreen = do
8   putStrLn "Is green your favorite color?"
9   isYes <$> getLine

```

与原文不同之处在于使用了 `<$>`，即 `fmap` 函数，替换掉了之前的 `inpStr <- getLine` 以及 `return`。

最后是一个做作的例子用于展示 `return` 并不需要位于 `do` 代码块的结尾。

```
1 returnTest :: IO ()
2 returnTest =
3     do one <- return 1
4         let two = 2
5         putStrLn $ show (one + two)
```

Haskell 真的是命令式的吗？

这些 `do` 代码块看起来像是命令式的语言。毕竟大多数情况下我们都是在按顺序运行命令。

但是 Haskell 本质上仍然是一种惰性语言。虽然有时有必要对 I/O 操作进行排序，但这是使用了 Haskell 中的工具来完成的。Haskell 还通过 IO 单子实现了 I/O 与语言其他部分的良好分离。

Lazy I/O 的副作用

本章提到的 `hGetContents`，其返回的 `String` 可以用在纯代码中。

我们需要更具体的了解副作用是什么。当我们说 Haskell 没有副作用时，这到底是什么意思？

某种层度上，副作用总是可能的。一个没有优化的循环，即使是用纯代码写的，也有可能导致系统的内存耗尽然后导致程序崩溃。或者它可能导致数据被交换到磁盘。

当我们提到没有副作用，意思其实是 Haskell 的纯代码不会运行能触发副作用的命令。纯函数不会修改一个全局变量，I/O 请求，或者运行一个命令关闭系统。

当你有一个 `hGetContents` 而来的字符串被传递至一个纯函数时，函数不会知道 `String` 是从磁盘文件而来的。它只会表现得跟通常那样，但处理该 `String` 可能会导致环境发出 I/O 命令。纯函数不会这么做；它们是纯函数处理的结果，就像将内存交换到磁盘的例子一样。

在某些情况下，你可能需要更多的控制 I/O 发生的确切时间。也许是正在交互式的从用户那里读取数据，或者通过管道从另一个程序读取数据，并需要直接与用户通信。这些情况下，`hGetContents` 可能不合适。

缓存

缓存模式

Haskell 有三种不同的缓存模式：`BufferMode` 类型：`NoBuffering`，`LineBuffering` 以及 `BlockBuffering`。

`NoBuffering` 正如其名 – 无缓存。通过例如 `hGetLine` 等函数读取的数据将每次从操作系统重读取一个字符。写入的数据将立刻写入，也经常一次写入一个字符。通常 `NoBuffering` 的性能很差，不适合通用用途。

`LineBuffering` 使输出缓冲区在输出换行符或缓存过大时被写入。在输入时，它通常会尝试读取任何可用的数据块，直到它第一次看到换行符。当从终端读取数据时，每次按下回车键后应该立刻返回数据。这通常是合理的默认值。

`BlockBuffering` 使 Haskell 在可能的情况下以固定大小块来读写数据。在批量处理大量数据时，这是性能最好的方法，即使这些数据是以行记录的。然而它不能用于交互式程序，因为它会阻塞输入，直到读取到完整的块。`BlockBuffering` 接受 `Maybe` 类型的参数：如果 `Nothing`，它将使用定义好的缓存大小；或者你可以使用 `Just 4096` 之类的设置将缓存设置为 4096 字节。

默认和缓存模式取决于操作系统和 Haskell 的实现。可以通过调用 `hGetBuffering` 向系统询问当前的缓存模式。当前的缓存模式可以通过 `hSetBuffering` 设置，它接受 `Handle` 和 `BufferMode`。例如 `hSetBuffering stdin (BlockBuffering Nothing)`。

刷新缓存

对于任何类型的缓存，有时我们会希望强制 Haskell 写出保存在缓存中的任何数据。有些时候这会自动发生：比如说调用 `hClose` 时。而调用 `hFlush` 将强制立刻写入任何挂起的数据。当 `Handle` 是一个网络 socket 且想要立刻传输数据时，或者是想要将磁盘上的数据提供给可能并发读取它的其它程序时，这会很有用。

读取命令行参数

`System.Environment.getArgs` 以 `IO [String]` 返回所有参数。类似 C 里的 `argv`，始于 `argv[1]`。程序名 (C 中的 `argv[0]`) 则可以通过 `System.Environment.getProgName` 得到。

`System.Console.GetOpt` 模块提供了一些解析命令行选项的工具。如果我们的程序有一些复杂的选项，该模块则很有帮助。

环境变量

读取环境变量可以用 `System.Environment` 的两个函数：`getEnv` 或 `getEnvironment`。前者查找特定变量，变量不存在时抛出异常；后者将所有环境变量以 `[(String, String)]` 输出，可以使用例如 `lookup` 函数来找到所需的变量。

8 Efficient file processing, regular expressions, and file name matching

高效文件处理

下面是一个简单的微基准测试，读取一个全是数字的文本文件，打印它们的总和：

```
1 main :: IO ()
2 main = do
3     contents <- getContents
4     print $ sumFile contents
5     where
6         sumFile = sum . map read . words
```

尽管 `String` 是用于读写文件的默认类型，它并不高效，因此像这样一个简单的程序性能是很差的。

`String` 代表着 `Char` 值的列表；列表中每个元素的内存分配都是独立的，需要额外的开销。这些因素会影响必须读写文本或二进制数据程序的内存消耗和性能。像这样的简单基准测试，即使是用解释性语言（如 Python）编写的程序，性能也比使用 `String` 的 Haskell 代码高出一个数量级。

`bytestring` 库提供了一个快速并且又低开销的 `String` 类型替代品。通过 `bytestring` 编写的代码通常可以媲美或超过 C 语言的性能和内存消耗，同时还保持了 Haskell 的表现力和简洁性。

该库提供了两个模块。每个定义的函数几乎都是 `String` 对应函数的直接替代品：

- `Data.ByteString` 模块定义了一个严格 *strict* 类型的 `ByteString`。其以二进制或文本数据的数组形式来表示字符串。
- `Data.ByteString.Lazy` 模块提供了惰性 *lazy* 类型的 `ByteString`。其以块 *chunks* 数组来表示字符串，最大大小为 64KB。

两个 `ByteString` 类型在特定环境都有更好的表现。对于流式处理一个大数（几百 MB 至 TB），惰性 `ByteString` 类型表现的最好。它的块大小被调整为适合现代 CPU 的 L1 缓存，且垃圾回收可以快速丢弃不再使用的流数据块。

而严格的 `ByteString` 类型在不太关心内存占用或需要随机访问数据的环境下性能更好。

二进制 I/O 与 qualified imports

现在来开发一个函数来说明 `ByteString` 的一些 API。我们将决定一个文件是否为 ELF 对象：这种格式多用于现代 Unix 类型系统的可执行文件。

查看文件的首四个字节，检查它们是否匹配一个字节列表：

```

1 import Data.ByteString.Lazy qualified as L
2
3 hasElfMagic :: L.ByteString -> Bool
4 hasElfMagic content = L.take 4 content == elfMagic
5 where
6     elfMagic = L.pack [0x7f, 0x45, 0x4c, 0x46]

```

`ByteString` 的惰性模块以及严格模块意图解决二进制 I/O。Haskell 用 `Word8` 来展示字节的数据类型；可以通过 `Data.Word` 来引入它。

上述例子的 `L.pack` 函数接受一个 `Word8` 列表，打包它们成为一个惰性的 `ByteString`。（`L.unpack` 函数作用相反。）而 `hasElfMagic` 函数则是比较 `ByteString` 的前四个字符。

以下是将其运用在一个文件上的例子：

```

1 isElfFile :: FilePath -> IO Bool
2 isElfFile path = do
3     content <- L.readFile path
4     return $ hasElfMagic content

```

`L.readFile` 函数是 `readFile` 的惰性 `ByteString` 版本，即按需读取。它效率很高，每次读取最大 64KB。惰性 `ByteString` 很适合我们的任务：因为我们至多读取文件的前四个字节，因此我们可以安全的使用该函数在任何大小的文件上。

Text I/O

`bytestring` 库还提供了两个模块用于 text I/O 功能，`Data.ByteString.Char8` 以及 `Data.ByteString.Lazy.Char8`。它们将单独的字符串元素导出成 `Char` 而不是 `Word8`。

Warning

上述模块中的函数进作用于字节大小的 `Char` 值，也就是仅适用于 ASCII 以及某些欧洲字符集。超过 255 的字符会被截断。

面向字符的 `bytestring` 模块为文本处理提供了一些有用的函数。以下是包含了月度的股票价格的文件：

```

1 ghci> putStr =<< readFile "prices.csv"
2 Date,Open,High,Low,Close,Volume,Adj Close
3 2008-08-01,20.09,20.12,19.53,19.80,19777000,19.80
4 2008-06-30,21.12,21.20,20.60,20.66,17173500,20.66
5 2008-05-30,27.07,27.10,26.63,26.76,17754100,26.76
6 2008-04-30,27.17,27.78,26.76,27.41,30597400,27.41

```

如何找到最高价呢？收盘价在第四列，用都好分隔。下面是得到收盘价的函数：

```

1 closing = readPrice . (!! 4) . L.split ','

```

该函数是 point-free 风格，需要从右往左阅读。`L.split` 将一个惰性的 `ByteString` 分隔开来，分隔发生在每一次找到匹配值时。`(!!)` 操作符则是获取列表中第 `k` 个元素。`readPrice` 函数将一个代表分数的字符串转换为数值：

```
1 readPrice :: L.ByteString -> Maybe Int
2 readPrice str = case L.readInt str of
3   Nothing -> Nothing
4   Just (dollars, rest) ->
5     case L.readInt (L.tail rest) of
6       Nothing -> Nothing
7       Just (cents, more) ->
8         Just (dollars * 100 + cents)
```

上面使用了 `L.readInt` 函数，即解析整数，返回整数以及字符串剩余部分。
找到最高收盘价的函数：

```
1 highestClose = maximum . (Nothing :) . map closing . L.lines
2
3 highestCloseFrom path = do
4   contents <- L.readFile path
5   print $ highestClose contents
```

这里使用了一个小技巧。当我们提供一个空列表给 `maximum` 函数会抛出异常：

```
1 ghci> maximum [3,6,2,9]
2 9
3 ghci> maximum []
4 *** Exception: Prelude.maximum: empty list
```

测试：

```
1 ghci> :load HighestClose
2 [1 of 1] Compiling Main             ( HighestClose.hs, interpreted )
3 Ok, modules loaded: Main.
4 ghci> highestCloseFrom "prices.csv"
5 Loading package array-0.1.0.0 ... linking ... done.
6 Loading package bytestring-0.9.0.1 ... linking ... done.
7 Just 2741
```

因为从逻辑中分离了 I/O，测试空数据时不需要创建一个空文件：

```
1 ghci> highestClose L.empty
2 Nothing
```

文件名匹配

很多面向系统的编程语言都提供了根据模式来匹配文件的功能。尽管 Haskell 的标准库有不错的系统编程工具，它并没有提供这类的匹配函数。

这类模式通常被称为 `glob` 模式，通配符模式或者 `shell` 样式模式。它们有一些简单的规则：

- 将字符串与模式进行匹配，从字符串的开头开始，到末尾结束。
- 大多字面字符都匹配自己。例如，模式中的文本 `foo` 将匹配输入字符串中的 `foo`，且仅匹配 `foo`。
- `*`（星号）字符表示“匹配任何内容”；它将匹配任何文本，包括空字符串。
- `?`（问号）字符匹配任何单个字符。模式 `pic?? .jpg` 将匹配 `picaa.jpg` 或 `pic01.jpg` 等名称。
- `[`（开方括号）字符开始一个字符类，以 `]` 结束。它的意思是“匹配这个类中的任何字符”。字符类可以在 `[` 后面加一个 `!` 来表示“匹配不属于这个类的任何字符”。

作为一种简写，一个字符后面跟着一个 `-`（破折号），后面跟着另一个字符，表示一个范围：“匹配此集合中的任何字符”

虽然 Haskell 的标准库中没有提供 `glob` 模式的匹配，但是它提供了一个非常好的正则表达式库。

Haskell 中的正则表达式

首先让我们加载 `Text.Regex.Posix` 库

```
1 ghci> :module +Text.Regex.Posix
```

根据官方 Wiki 提到的 `regex-posix` 速度非常的慢，在生产代码中需要换成别的库。

我们这里只需要正则表达式匹配的函数，一个中缀操作符 `(=~)` 即可（从 Perl 中借用）。要克服的第一个障碍是 Haskell 的 `regex` 库大量使用了多态。因此 `(=~)` 操作符的类型签名很难理解，因此这里暂时不做解释。

`=~` 对它的两个参数和返回类型使用了 `typeclasses`。第一个参数（即 `=~` 左侧）是要匹配的文本；第二个参数（右侧）则是要匹配的正则表达式。我们可以传 `String` 或 `ByteString` 作为参数。

结果的多种类型

`=~` 操作符的结果类型是多态的，因此 Haskell 编译器需要知道其类型。在真实代码中，它有可能根据使用的场景被推导出正确的类型。但是 `ghci` 没法这么做，它没有足够的信息来进行推导。

当 `ghci` 不能推导 `target` 类型时，我们需要进行显式指定：

```

1 ghci> "my left foot" =~ "foo" :: Bool
2 Loading package array-0.1.0.0 ... linking ... done.
3 Loading package containers-0.1.0.1 ... linking ... done.
4 Loading package bytestring-0.9.0.1 ... linking ... done.
5 Loading package mtl-1.1.0.0 ... linking ... done.
6 Loading package regex-base-0.93.1 ... linking ... done.
7 Loading package regex-posix-0.93.1 ... linking ... done.
8 True
9 ghci> "your right hand" =~ "bar" :: Bool
10 False
11 ghci> "your right hand" =~ "(hand|foot)" :: Bool
12 True

```

正则表达式库的核心是一个名为 `RegexContext` 的 typeclass，用于描述 `target` 类型的行为；基础库为我们定义了很多实例。`Bool` 类型就定义了该 typeclass 的实例，因此我们能得到可用的结果。另一个实例就是 `Int`：

```

1 ghci> "a star called henry" =~ "planet" :: Int
2 0
3 ghci> "honorificabilitudinitatibus" =~ "[aeiou]" :: Int
4 13

```

如果寻求的是 `String` 结果，我们将得到第一个匹配的子字符串，或是不匹配时得到一个空字符串。

```

1 ghci> "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: String
2 "ii"
3 ghci> "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: String
4 ""

```

另外一个可作为结果的类型是 `[String]`，其返回所有匹配字符串的列表：

```

1 ghci> "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: [String]
2
3 <interactive>:1:0:
4   No instance for (RegexContext Regex [Char] [String])
5     arising from a use of `=~' at <interactive>:1:0-50
6   Possible fix:
7     add an instance declaration for
8     (RegexContext Regex [Char] [String])
9   In the expression:
10     "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: [String]
11   In the definition of `it':
12     it = "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" ::
13         [String]
14 ghci> "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: [String]
15
16 <interactive>:1:0:
17   No instance for (RegexContext Regex [Char] [String])
18     arising from a use of `=~' at <interactive>:1:0-54

```

```

19     Possible fix:
20         add an instance declaration for
21         (RegexContext Regex [Char] [String])
22     In the expression:
23         "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: [String]
24     In the definition of `it`:
25         it = "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" ::
26             [String]

```

Tip

注意字符串返回

如果期望结果是一个 `String`，那么就要小心了。因为 `(=~)` 返回一个空字符串来表示“没有匹配”，如果空字符串也可能是 `regex` 的有效匹配，那么就会难以识别。在这种情况下就应该是有一个不同的返回类型，例如 `[String]`。

上述都是“简单的”返回类型，这远没有结束。继续往下推进前，让我们定义一个在之后例子中都会用到的匹配：

```

1 ghci> let pat = "(foo[a-z]*bar|quux)"

```

我们可以获得关于匹配发生的上下文的大量信息。如果期望是 `(String,String,String)` 元组，则会在第一个匹配之前获得文本，匹配成功的文本，以及紧随其后的文本。

```

1 ghci> "before foodiebar after" =~ pat :: (String,String,String)
2 ("before ", "foodiebar", " after")

```

如果匹配失败，那么整个文本则会在元组中以“之前”元素作为返回，而其它两个元素为空：

```

1 ghci> "no match here" =~ pat :: (String,String,String)
2 ("no match here", "", "")

```

请求一个四元素元组则会给我们第四个元素，它是匹配模式中所有组的列表：

```

1 ghci> "before foodiebar after" =~ pat :: (String,String,String,[String])
2 ("before ", "foodiebar", " after", ["foodiebar"])

```

我们还可以获取匹配的数值信息。一对 `Int` 给出的第一个匹配的起始偏移量以及长度。如果我们请求这些配对的列表，我们将得到所有匹配的信息。

```

1 ghci> "before foodiebar after" =~ pat :: (Int,Int)
2 (7,9)
3 ghci> "i foobarbar a quux" =~ pat :: [(Int,Int)]
4
5 <interactive>:1:0:
6   No instance for (RegexContext Regex [Char] [(Int, Int)])
7   arising from a use of `=~' at <interactive>:1:0-26
8   Possible fix:

```



```

9      add an instance declaration for
10      (RegexContext Regex [Char] [(Int, Int)])
11      In the expression: "i foobarbar a quux" =~ pat :: [(Int, Int)]
12      In the definition of `it`:
13      it = "i foobarbar a quux" =~ pat :: [(Int, Int)]

```

如果请求单个元组，则用值 `-1` 作为元组的第一个元素（匹配偏移量）表示匹配失败；如果请求元组列表，则用空列表表示匹配失败：

```

1  ghci> "eleemosynary" =~ pat :: (Int,Int)
2  (-1,0)
3  ghci> "mondegreen" =~ pat :: [(Int,Int)]
4
5  <interactive>:1:0:
6    No instance for (RegexContext Regex [Char] [(Int, Int)])
7    arising from a use of `=~` at <interactive>:1:0-18
8    Possible fix:
9      add an instance declaration for
10      (RegexContext Regex [Char] [(Int, Int)])
11      In the expression: "mondegreen" =~ pat :: [(Int, Int)]
12      In the definition of `it`: it = "mondegreen" =~ pat :: [(Int, Int)]

```

更多的关于正则表达式

混合与匹配字符串类型

之前提到 `=~` 操作符使用 `typeclasses` 作为其参数类型以及返回类型。可以使用 `String` 或严格 `ByteString` 的值来进行正则表达式计算。

```

1  ghci> :module +Data.ByteString.Char8
2  ghci> :type pack "foo"
3  pack "foo" :: ByteString

```

可以尝试不同组合的 `String` 与 `ByteString`：

```

1  ghci> pack "foo" =~ "bar" :: Bool
2  False
3  ghci> "foo" =~ pack "bar" :: Int
4  0
5  ghci> pack "foo" =~ pack "o" :: [(Int, Int)]
6
7  <interactive>:1:0:
8    No instance for (RegexContext Regex ByteString [(Int, Int)])
9    arising from a use of `=~` at <interactive>:1:0-21
10   Possible fix:
11     add an instance declaration for
12     (RegexContext Regex ByteString [(Int, Int)])
13     In the expression: pack "foo" =~ pack "o" :: [(Int, Int)]

```

```

14     In the definition of `it':
15     it = pack "foo" =~ pack "o" :: [(Int, Int)]

```

不过需要注意的是，如果期望匹配结果中包含字符串值，那么所匹配的文本必须是同一类型的字符串：

```

1  ghci> pack "good food" =~ ".ood" :: [ByteString]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex ByteString [ByteString])
5      arising from a use of `=~' at <interactive>:1:0-25
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex ByteString [ByteString])
9      In the expression: pack "good food" =~ ".ood" :: [ByteString]
10     In the definition of `it':
11     it = pack "good food" =~ ".ood" :: [ByteString]

```

上述例子中，使用 `pack` 将 `String` 转为了 `ByteString`。类型检查器可以接受是因为 `ByteString` 在结果类型中展示出来了。但是如果尝试获取一个 `String`，则不会成功：

```

1  ghci> "good food" =~ ".ood" :: [ByteString]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex [Char] [ByteString])
5      arising from a use of `=~' at <interactive>:1:0-20
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex [Char] [ByteString])
9      In the expression: "good food" =~ ".ood" :: [ByteString]
10     In the definition of `it':
11     it = "good food" =~ ".ood" :: [ByteString]

```

我们可以通过将左侧的字符串类型与结果再次匹配来简单的修复这个问题：

```

1  ghci> "good food" =~ ".ood" :: [String]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex [Char] [String])
5      arising from a use of `=~' at <interactive>:1:0-20
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex [Char] [String])
9      In the expression: "good food" =~ ".ood" :: [String]
10     In the definition of `it': it = "good food" =~ ".ood" :: [String]

```

其它需要注意的事项

查看 Haskell 库文档时，可以看到几个与 `regex` 相关的模块。`Text.Regex.Base` 模块定义了所有其他 `regex` 模块所遵循的通用 API。可以同时安装 `regex` API 的多个实现。在编写本文时，GHC 与一个实现 `Text.Regex.Posix` 捆绑在一起。顾名思义，这个包提供 POSIX `regex` 语义。

Tip

Perl 与 POSIX 正则表达式

如果是从 Perl、Python 或 Java 等语言来学习 Haskell 的，并且在其中一种语言中使用过正则表达式，那么应该意识到由 `Text.Regex.Posix` 模块处理的 POSIX `regex` 与 Perl 风格的 `regex` 在某些重要方面是不同的。以下是一些比较明显的区别。

Perl `regex` 引擎在匹配选项时执行左偏匹配，而 POSIX 引擎选择最贪婪的匹配。这意味着给定一个正则表达式 `(foo|fo*)` 和一个文本字符串 `foooooo`，perl 风格的引擎将给出 `foo` 的匹配（最左边的匹配），而 POSIX 引擎将匹配整个字符串（最贪婪的匹配）。

POSIX `regex` 的语法不如 perl 风格的 `regex` 统一。它们还缺乏 perl 风格的 `regex` 提供的许多功能，例如零宽度断言和对贪婪匹配的控制。

将一个 glob 模式转为一个正则表达式

我们已经见识到了很多种将文本域正则表达式匹配的方法，让我们将注意力转回到 glob 模式。我们希望编写一个接受一个 glob 模式的函数，将其表示形式返回为正则表达式。glob 模式和正则表达式两者都是 `text` 字符串，因此函数的类型就很明确了：

```
1 module GlobRegex
2   ( globToRegex,
3   )
4 where
5
6 import Text.Regex.Posix ((=))
7
8 globToRegex :: String -> String
```

我们生成的正则表达式必须是固定的，使得它可以从同到尾的匹配字符串。

```
1 globToRegex cs = '^' : globToRegex' cs ++ "$"
```

由于 `String` 就是 `[Char]` 的别名，这里 `:` 操作符将一个值（这里是 `^` 字符）放到一个列表的头部，而这个列表则是还未出现的 `globToRegex` 函数所返回的。

Tip

在定义一个值之前使用它

Haskell 使用一个值或者函数时，并不需要它们被声明或者在一个源文件中定义。一个定义出现在第一个被使用的地方之后是很正常的。Haskell 编译器并不关心这一层面上的顺序。这使我们可以灵活的按照逻辑来构建代码，而不是遵循编译器编写者的顺序。

Haskell 模块的编写者通常会使用这个灵活性来让“更加重要”的代码前置置于源文件中。这正式我们编写 `globToRegex` 函数以及其帮助函数的方式。

有了正则表达式以后 `globToRegex` 函数则用作于大部分的翻译工作。以下使用 Haskell 的模式匹配来进行编码：

```
1 globToRegex' :: String -> String
2 globToRegex' "" = ""
3 globToRegex' ('*' : cs) = ".*" ++ globToRegex' cs
4 globToRegex' ('?' : cs) = '.' : globToRegex' cs
5 globToRegex' ('[' : '!' : c : cs) = "[" ++ c : charClass cs
6 globToRegex' ('[' : c : cs) = '[' : c : charClass cs
7 globToRegex' ('[' : _) = error "unterminated character class"
8 globToRegex' (c : cs) = escape c ++ globToRegex' cs
```

第一个子句规定了如果达到了 glob 模式的末尾（到那时会看到空字符串），返回 `$`，即正则表达式中的“匹配到了末尾”。接下来的是的子句则是一系列从 glob 语义切换到正则表达语义的模式。最后一个子句处理其它的每个字符，转义优先。

`escape` 函数确保正则引擎不会解释某些特定字符：

```
1 escape :: Char -> String
2 escape c
3   | c `elem` regexChars = '\\' : [c]
4   | otherwise = [c]
5 where
6   regexChars = "\\+()~$.{}|!"
```

`charClass` 帮助函数用于检测字符是否正确的终止，不修改输入直到遇到 `]`：

```
1 charClass :: String -> String
2 charClass (']' : cs) = ']' : globToRegex' cs
3 charClass (c : cs) = c : charClass cs
4 charClass [] = error "unterminated character class"
```

我们已经定义好了 `globToRegex` 与其帮助函数们，现在加载进 `ghci` 中测试一下：

```
1 ghci> :set -package regex-posix
2 package flags have changed, resetting and loading new packages...
3 ghci> :l GlobRegex.hs
4 [1 of 1] Compiling GlobRegex      ( GlobRegex.hs, interpreted )
```

```

5 Ok, one module loaded.
6 ghci> globToRegex "f??.c"
7 "^f\\.\\.\\.c$"

```

看起来像是一个合理的正则，那么用它来匹配字符串呢？

```

1 ghci> "foo.c" =~ globToRegex "f??.c" :: Bool
2 True
3 ghci> "test.c" =~ globToRegex "t[ea]s*" :: Bool
4 True
5 ghci> "teste.txt" =~ globToRegex "t[ea]s*" :: Bool
6 True

```

正常工作！我们还可以为 `fnmatch` 创建一个临时定义：

```

1 ghci> let fnmatch pat name = name =~ globToRegex pat :: Bool
2 ghci> :type fnmatch
3 fnmatch
4   :: rgx-bs-0.94.0.2-01560e7d:Text.Regex.Base.RegexLike.RegexLike
5     Text.Regex.Posix.Wrap.Regex source1 =>
6     String -> source1 -> Bool
7 ghci> fnmatch "d*" "myname"
8 False

```

`fnmatch` 并不符合“Haskell 本性”，通常 Haskell 风格的函数都需要描述性的“驼峰”名称，例如“file name matches”为 `fileNameMatches`，那么我们的库可以这样命名：

```

1 matchesGlob :: FilePath -> String -> Bool
2 name `matchesGlob` pat = name =~ globToRegex pat

```

一个重要的旁白：编写惰性函数

在一个命令式的语言中，`globToRegex'` 函数通常会表述为一个循环。例如 Python 的标准 `fnmatch` 模块中包含了一个名为 `translate` 的函数，其功能与我们的 `globToRegex` 函数一致。它的实现就是循环。

如果你有别的函数式编程的经验例如 Scheme 或 ML，那么肯定有“模仿循环的方法就是通过尾递归”这个概念。

再来看一下 `globToRegex'` 函数，并不是一个尾递归函数。可以测试一下最后的子句（其它子句结构都类似）。

```

1 globToRegex' (c : cs) = escape c ++ globToRegex' cs

```

它递归的应用了自身，且递归的结果作为一个参数在 `(++)` 函数上。由于递归的应用并非作为函数的最后处理部分，那么 `globToRegex'` 并不是一个尾递归。

那么为什么这个函数并非尾递归呢？答案就在 Haskell 的非严格计算策略。在提及这个话题之前，让我们快速的了解为什么在传统语言中，我们需要避免这样的尾递归定义。以下是 `(++)` 操作符的一个简单定义。它属于递归，但不是尾递归。

```

1  (++) :: [a] -> [a] -> [a]
2
3  (x:xs) ++ ys = x : (xs ++ ys)
4  []      ++ ys = ys

```

在严格执行语言中，如果我们计算 `"foo" ++ "bar"`，整个列表会被构建出来，接着再返回。非严格计算则会延迟直到真正需要被用到的时候。

如果需要表达式 `"foo" + "bar"` 的一个元素时，函数定义的第一个模式将被匹配，接着返回表达式 `x : (xs ++ ys)`。因为 `(:)` 构造函数是非严格的，那么 `xs ++ ys` 的计算就能被递延：按需生产更多的元素。当我们生产更多的结果时，我们将不再使用 `x`，因此垃圾回收器将会进行回收。由于是按需生产元素，且不持有已经完成的部分，编译器就可以以常量的空间来计算我们的代码。

使用我们的模式匹配器

通过一个函数用于匹配 glob 模式是很好的，但是我们希望将其用于实际应用。在 Unix 系的系统中 `glob` 函数返回所有匹配到的文件名称以及路径。让我们在 Haskell 中也构建一个类似的函数：

```

1  module Glob (namesMatching) where

```

`System.Directory` 模块提供了用于处理路径以及内容的标准函数。

```

1  import System.Directory (doesDirectoryExist, doesFileExist, getCurrentDirectory,
    getDirectoryContents)

```

而 `System.FilePath` 模块抽象了操作系统路径名称转换的细节，`(</>)` 函数将两个路径连接在一起：

```

1  ghci> :m +System.FilePath
2  ghci> "foo" </> "bar"
3  "foo/bar"

```

`dropTrailingPathSeparator` 函数正如其名，去除末尾的路径分隔：

```

1  ghci> dropTrailingPathSeparator "foo/"
2  "foo"

```

`splitFileName` 则分离了路径与文件：

```

1  ghci> splitFileName "foo/bar/Quux.hs"
2  ("foo/bar/", "Quux.hs")

```

`System.Directory` 模块结合 `System.FilePath` 的使用，我们可以编写一个同时作用于 Unix 系与 Windows 的 `namesMatching` 函数：

```

1  import System.FilePath (dropTrailingPathSeparator, splitFileName, (</>))

```

在这个模块中，我们将模仿一个“for”循环；首次在 Haskell 中尝试错误处理；并使用写好的 `matchesGlob` 函数。

```
1 import Control.Exception (handle)
2 import Control.Monad (forM)
3 import GlobRegex (matchesGlob)
```

由于路径与文件存在于拥有副作用的“真实世界”，我们的 globbing 函数的返回值则必须带上 `IO`。

如果传入的字符串没有包含模式字符，那么就简单的检查给定的名称是否存在于文件系统重。（注意，我们在这里使用 Haskell 的函数守护语法来编写一个漂亮整洁的定义。使用“if”也是可以的，不过就没有这么美观了）

```
1 isPattern :: String -> Bool
2 isPattern = any (`elem` "[*?")
3
4 namesMatching pat
5   | not (isPattern pat) = do
6     exists <- doesNameExist pat
7     return ([pat | exists])
```

函数 `doesNameExist` 会在稍后进行定义。

那么如果字符串是一个 glob 模式呢？继续我们的函数定义：

```
1 | otherwise = do
2   case splitFileName pat of
3     ("", baseName) -> do
4       curDir <- getCurrentDirectory
5       listMatches curDir baseName
6     (dirName, baseName) -> do
7       dirs <-
8         if isPattern dirName
9           then namesMatching (dropTrailingPathSeparator dirName)
10          else return [dirName]
11       let listDir =
12         if isPattern baseName
13           then listMatches
14           else listPlain
15       pathNames <- forM dirs $ \dir -> do
16         baseNames <- listDir dir baseName
17         return (map (dir </>) baseNames)
18       return (concat pathNames)
```

我们使用了 `splitFileName` 将字符串分离成了一对“所有东西除了最终名称”以及“最终名称”。如果首个元素为空，则在当前路径寻找一个模式；否则检查路径名称是否包含模式，没有包含则创建一个包含路径名称的单例列表，包含则列举所有匹配的路径。

Tip

需要注意的事情

`System.FilePath` 模块可以变得有点棘手。上述的例子中 `splitFileName` 函数留下了末尾的斜杠

```
1 ghci> :module +System.FilePath
2 ghci> splitFileName "foo/bar"
3 ("foo/", "bar")
4
```

如果我们忘了（或者不了解）要移除斜杠，我们则会无限在 `namesMatching` 中递归，因为下列行为

```
1 ghci> splitFileName "foo/"
2 ("foo/", "")
3
```

你可以猜想一下之后将会发生什么！

最后，收集了所有匹配的路径，得到了一个列表的列表，再将它们打平成一个单独的列表。

这里陌生的 `forM` 函数其行为类似于一个“for”循环：它映射它的第二个参数（一个 action）在第一个参数（一个列表）上，并返回一个列表的结果。

还剩下一些需要处理的问题。首先是 `doesNameExist` 函数。`System.Directory` 模块并不能检查一个名称是否存在与文件系统中，它强制要求我们决定检查的是文件还是路径。这个 API 很丑陋，因此我们需要将两者合二为一。首先检查是否为文件，因为文件相较于路径更为普遍。

```
1 doesNameExist :: FilePath -> IO Bool
2 doesNameExist name = do
3   fileExists <- doesFileExist name
4   if fileExists
5     then return True
6     else doesDirectoryExist name
```

还有另外两个函数需要定义，它们的返回都是一个路径中的所有名称。`listMatches` 函数返回的是一个路径中匹配了给定 glob 模式的所有文件。

```
1 listMatches :: FilePath -> String -> IO [String]
2 listMatches dirName pat = do
3   dirName' <-
4     if null dirName
5       then getCurrentDirectory
6       else return dirName
7   handle errorHandler $ do
8     names <- getDirectoryContents dirName'
```



```

9     let names' =
10         if isHidden pat
11         then filter isHidden names
12         else filter (not . isHidden) names
13     return (filter (`matchesGlob` pat) names')
14 where
15     errorHandler :: SomeException -> IO [String]
16     errorHandler _ = return []
17
18 isHidden :: [Char] -> Bool
19 isHidden ('.' : _) = True
20 isHidden _ = False

```

与原文不同之处在于 `handle errorHandler` 这里需要显式的为 `errorHandler` 标注类型，否则原文的 `handle (const (return []))` 将会在编译时抛出类型不明确的异常，详见StackOverflow。

`listPlain` 函数的返回要么是一个空的列表要么是一个单例列表，这取决于传递的名称是否存在：

```

1 listPlain :: FilePath -> String -> IO [String]
2 listPlain dirName baseName = do
3     exists <-
4         if null baseName
5         then doesDirectoryExist dirName
6         else doesNameExist (dirName </> baseName)
7     return ([baseName | exists])

```

通过 API 设计来处理错误

如果传递给我们的 `globToRegex` 的是一个错误的模式，并不会是一场灾难。这有可能使用户输入了错误的模式，这种情况下我们需要报告有意义的错误信息。

```

1 type GlobError = String
2
3 globToRegex :: String -> Either GlobError String

```

运行我们的代码

`namesMatching` 函数并不能很好地单独工作，我们需要将其与其它函数结合。

首先定义一个 `renameWith` 函数，并非只是简单的为一个文件重命名，而是将一个函数应用至文件名称，用函数返回的值来重命名文件。

```

1 import System.FilePath (replaceExtensions)
2 import System.Directory (doesFileExist, renameDirectory, renameFile)
3 import Glob (namesMatching)

```

```

4
5 renameWith :: (FilePath -> FilePath) -> FilePath -> IO FilePath
6 renameWith f path = do
7   let path' = f path
8   rename path path'
9   return path'

```

这里还是需要一个帮助函数来处理文件与路径：

```

1 rename :: FilePath -> FilePath -> IO ()
2 rename old new = do
3   isFile <- doesFileExist old
4   let f = if isFile then renameFile else renameDirectory
5   f old new

```

`System.FilePath` 模块提供了很多操作文件名的函数，它们可以很好地与 `renameWith` 与 `namesMatching` 函数结合，使得我们可以快速的使用它们来创造复杂行为的函数。例如修改 C++ 源文件的后缀。

```

1 cc2cpp :: IO [FilePath]
2 cc2cpp = mapM (renameWith (flip replaceExtension ".cpp")) =<< namesMatching "*.cc"

```

`flip` 函数接受另一个函数作为参数，交换该函数的入参。`=<<` 函数将其右侧 action 的结果喂给左侧的 action。

9 I/O case study: a library for searching the filesystem

WIP

10 Code case study: parsing a binary data format

WIP

11 Testing and quality assurance

WIP

12 Barcode recognition

WIP

13 Data structures

WIP

14 Monads

WIP

15 Programming with monads

WIP

16 The Parsec parsing library

WIP

17 The foreign function interface

WIP

18 Monad transformers

WIP

19 Error handling

WIP

20 Systems programming

WIP

21 Working with databases

WIP

22 Web client programming

WIP

23 GUI programming

WIP

24 Basic concurrent and parallel programming

WIP

25 Profiling and tuning for performance

WIP

26 Advanced library design: building a Bloom filter

WIP

27 Network programming

WIP

28 Software transactional memory

WIP